

بسم الله الرحمن الرحيم

تعلم كيفية التعامل مع الوينسوك

شرح / الاستاذ . رغيد الطيب
اعداد و تنسيق و تعديل / علي اللوماني



Modi Link Network

Copyright © 2003 by : Ali Al-Lomani

www.modilink.com



www.vb4arab.com/vb

ملاحظة : سوف تجد مع هذا الكتاب عدد من الامثلة لتسهيل الدروس عليك

هذه الدروس أخذت من :

<http://www.vb4arab.com/vb/showthread.php?threadid=17975>

مقدمة

السلام عليكم ورحمة الله وبركاته...

اخوتي الكرام قرأت الكثير من الاستفسارات عن الاداه Winsock وكنت حتى وقت قريب بل منذ اسابيع لا اعرف شيئاً عنها ولكن من الله علي بفهم بسيط عنها ... وسوف احاول هنا نقل كل ما لدي عن هذه الاداه اليك اخي على ان اساهم في رفع غموضها عنك اخي كما ساهم بعض الاخوة في رفع غموضها عني وبسم الله الرحمن الرحيم نبدء:

Winsock هي اداة تستخدم لربط برنامجين مع بعضهما البعض سواء كان هاذان البرنامجان في كمبيوتر واحد ام في كمبيوترين مختلفين وسواء في شبكة داخلية او خارجية

والغرض من هذا الربط هو نقل البيانات من طرف الى طرف آخر وقد يكون الغرض من وراء نقل البيانات هو غرض مفيد او غير مفيد كما التجسس على الآخرين....

ولكي نبدء بالبرمجة مع هذه الاداه يجب علينا اولاً ان نعرف ما هي الشروط الاساسية والاحتياجات لعملية الربط اهم هذه الشروط هو معرفة الاي بي IP وهو عبارة عن رقم يقوم مزود الانترنت لديك بمنحك اياه عند دخولك الشبكة وهو الرقم الذي يميز جهازك عن باقي الاجهزة داخل الشبكة ولا يمكن ان يتساوى جهازان في الاي بي عند اتصالهما وهو رقم غير ثابت اي انك اذا دخلت الشبكة يتم اعطائك رقم فاذا قمت بالخروج ثم العودة مرة اخرى ستجد ان هذا الرقم قد تغير ما يهنا معرفته هو ان هذا الرقم ضروري جداً للربط بين الاجهزة ويمكن الاستعاضة عنه في حالة الشبكة الداخلية باسم الجهاز الذي نريد الاتصال به وبشكل افتراضي اذا كنت غير مرتبط في الشبكة الان فان رقم الاي بي التابع لجهازك هو ١٢٧,٠,٠,١ وعند اتباطك فانه يتغير بحسب ما يعطى لك من قبل مزود الخدمة لديك

والشرط الثاني هو المنفذ او Port حيث يمكن للبرنامج استخدام اي منفذ الى اكثر من ستين الف منفذ وهو يستخدم للتراسل بين البرامج مثلاً اذا جعلت برنامج في جهاز A وآخر في جهاز B واردت ان يتصل البرنامج A بالبرنامج B يجب عليك اولاً ان تجعل البرنامج B ينتظر الاتصال على منفذ معين مثلاً المنفذ رقم ١٠٠٠٠ ثم تجعل البرنامج A يطلب الاتصال بالجهاز الذي يمتلك رقم IP التابع للبرنامج B فيقوم البرنامج بطلب الاتصال عن طريق الاداه السابقة الذكر ولكن قبل ذلك يجب ان تحدد المنفذ الذي تريد الاتصال به والذي يجب ان يكون ايضاً ١٠٠٠٠ وعندما يبدء البرنامج A الاتصال على المنفذ المذكور يجد ان البرنامج B منتظراً للاتصال على هذا المنفذ فيحصل بذلك الاتصال وعند هذه النقطة يمكن فعل الكثير....

فمثلاً في برامج التجسس نسمع كثيراً عن مصطلح السيرفر او الباتش وهو في مثالنا البرنامج الثاني حيث يقوم باختيار احد المنافذ ثم الانتظار على هذا المنفذ وعندما تتصل بالشبكة يعمل الهكر على معرفة الاي بي التابع لجهازك ثم يطلب الاتصال به على نفس المنفذ الذي ينتظره فيه برنامج السيرفر وبذلك يحصل الاتصال بينهما ثم يقوم حينها الهكر بسرقة كل البيانات التي يريد من جهازك ثم ارسالها الى البرنامج الآخر على جهازه

هذا هو مفهوم بسيط لفكرة الاتصال وسوف انطرق لكيفية تطبيق ذلك على شكل دروس بإذن الله تعالى

اعداد الأداة

فقد تعبت كثيراً حتى تكونت لدي فكرة بسيطة احاول ان انقلها لكم اخوتي بالتدريج بحيث نفهم الموضوع بشكل عام ونظري اولاً ثم نقوم بالبدء في البرمجة وعندها فقط اتق في انكم جميعاً سوف تنتجون برامج اجمل واقرى مما قمت به انا فلستم بحاجة الا الى وضع ارجلكم على اول الطريق لتتلقوا تسابقون الرياح وهذا ما اسعى اليه

بمعون الله تعالى....

ودعونا الان نبدء بالخطوة التالية في فهم اساسيات تفكيرنا بالبرنامج حيث اننا عندما نقوم بكتابة برنامج الترابط نقوم بتطوير برنامجان في وقت واحد وليس برنامج واحد احدهما Server او الباتش وهو الذي ينتظر الاتصال والآخر هو العميل او Client وهو الذي يقوم بطلب الاتصال وسوف نسميهما نحن A و B لسهولة استخدامها في الشرح

قبل كل شيء يجب ان نضيف الاداه Winsock الى برنامجنا وذلك عن طريق النقر على Components من القائمة Project ثم اختيار العنوان [Microsoft Winsock Control 6.0] من مربع الحوار الذي سيظهر وبذلك نكون قد اضعنا الاداة Winsock الى برنامجنا
واليكم الان بعض من الخواص المهمة في هذه الاداه والتي يجب علينا فهمها قبل كل شيء :

LocatPort	وهو يحدد رقم المنفذ المحلي
LocalIP	يعطي رقم الاي بي المحلي
RemotePort	رقم المنفذ في الجهاز الآخر
RemoteHost	رقم الاي بي للجهاز الآخر
Protocol	نوع الترابط بين البرنامجين ويوجد نوعان نتكلم عنهما لاحقاً
State	يعطي حالة البرنامج الان هل هو متصل ام لا ام ينتظر اتصال وهكذا

وكذلك توجد اوامر اخرى مهمة جداً مثل :

Listen	تستخدم لجعل البرنامج ينتظر الاتصال
GetData	تستخدم لقراءة البيانات التي ارسلت الى البرنامج
SendData	تستخدم لارسال البيانات الى البرنامج الآخر
Connect	تستخدم لطلب الاتصال
Accept	تستخدم لقبول الاتصال

وبيقى اخيراً مجموعة من الاحداث المهمة وهي:

Connect	يفيد بان عملية الربط قد تمت
Close	يفيد بان الربط مفقود
DataArrival	يعلم البرنامج بوصول بيانات من البرنامج الآخر
SendComplete	يفيد البرنامج ان الارسال قد انتهى
Error	ينبه بوجود خطأ ما
ConnectionRequest	يفيد بانه يوجد برنامج يريد الاتصال ببرنامجنا

طبعاً يبدو الان الامر قد تعقد وانه يجب علينا حفظ الكثر حتى نستخدم الاداه ولكني اريد أن أعلمكم ان الامر ليس كذلك فانت لكي تقوم بعملية ربط برنامجين ببعض يلزمك فقط سطر واحد من الشفرة في البرنامج الاول وسطران في البرنامج الآخر فقط !!! طبعاً هذا يفيد بعملية الربط ويزيد الكود قليلاً اذا اردت ان ترسل رسالة من البرنامج الاول الى البرنامج الآخر لذى لا اريد ان تصاب اخي بخيبة امل من الكثرة التي تجدها هنا لأنني ذكرتها فقط حتى اذا استخدمتها لاحقاً امكنك ان تعود الى هذه البيانات لمعرفة الغرض الاساسي من الامر اي فقط كمرجع لذى من الان وصاعداً هيئ نفسك لنبدء اول عملية ربط بين برنامجينا وسوف اجعلها ببساطة جداً لكي يتحمس الذي لم يجرب ذلك من قبل واتمنى ان لا تحبط الذي قام بالربط مسبقاً لاننا سوف ندرج كي نصل الى برامج كبيرة بإذن الله وتوفيقه ...

الدرس الأول :

بسم الله الرحمن الرحيم

سوف نبدء بتصميم اول برنامج وهو عبارة عن برنامج يقوم بارسال رسالة من البرنامج (أ) الى البرنامج (ب) حيث ان البرنامج (أ) هو العميل والبرنامج (ب) هو السيرفر وكنا قد علمنا ان العميل هو الذي يقوم بالاتصال

والسيرفر ينتظر هذا الاتصال ...

تصميم البرنامج (أ) :

نقوم أولاً بفتح مشروع جديد ثم نضع عليه مربع نص نسميه txtMsg ونضيف زرّين الأول نجعل عنوانه اتصال والآخر ارسال ثم نضيف أخيراً اداه Winsock لكي نقوم بالاتصال ونجعل اسمها wsk

ثم نكتب الشفرة التالية للزر اتصال :

```
Wsk.Close  
Wsk.RemoteHost = "127.0.0.1"  
Wsk.RemotePort = 10000  
Wsk.Connect
```

في الشفرة السابقة قمنا أولاً باستدعاء الامر Close والذي يقوم بقطع الاتصال الحالي حيث انه اذا اردنا ان نبدء اتصال جديد وكان البرنامج اصلاً متصل سوف يسبب ذلك خطأ لذي نقوم أولاً بقطع الاتصال السابق ثم نطلب اتصال جديد ولقد عرفنا سابقاً اننا لكي نتصل بجهاز آخر يجب ان نعرف رقم الاي بي الخاص بذلك الجهاز ونضعه في الخاصية RemoteHost اي انني اريد الاتصال بالكمبيوتر الذي رقمه هو "١٢٧,٠,٠,١" وكنا قد علمنا ايضاً ان هذا الرقم "١٢٧,٠,٠,١" هو الرقم الافتراضي لجهازك اي اننا نخبر الاداه wsk بان تقوم بالاتصال بجهازك نفسه لان هذا هو رقمه بحيث يتسنى لنا التجربة ثم اذا اردنا بعد ذلك ان نتراسل مع جهاز آخر نكتب رقم الاي بي التابع له بدل هذا الرقم
ثم قمنا ايضاً بتحديد رقم المنفذ الذي نريد الاتصال به وهو في مثالنا ١٠٠٠٠ ثم اخيراً قمنا بطلب الاتصال بواسطة الامر Connect اذا يمكننا قرأه الاوامر التي ارسلنها الى الاداه wsk كالتالي :

"قم أولاً باغلاق الاتصال السابق ثم قم بالاتصال بالجهاز الذي يمتلك الرقم ١٢٧,٠,٠,١ على المنفذ ١٠٠٠٠"

يمكن ايضاً الاستعاضة عن السابق بالامر التالي اختصاراً :

```
wsk.Close  
wsk.Connect "127.0.0.1" , 10000
```

وسيقوم بنفس العمل السابق....

وبهذا نكون قد قمنا بتنفيذ كود الاتصال في برنامج العميل وبقي ان نمكن البرنامج من ارسال رسالة الى البرنامج (ب) بحيث نرسل النص الموجود في مربع النص txtMsg لذا نكتب الاوامر التالية في الزر ارسال :

```
wsk.SendData "" : txtMsg.Text  
DoEvents
```

علمنا سابقاً انه اذا اردنا ان نرسل شيئاً نستخدم الامر SendData وقد قمنا هنا بارسال العبارة "رسالة من أ : " ثم محتويات مربع النص ... txtMsg اما الامر DoEvents فهو يضع الفرصة للرسالة حتى يتم ارسالها اي يقوم بعملية انتظار انتهاء الارسال وبذلك نكون قد انتهينا من البرنامج الاول

تصميم السيرفر (ب) :

تختلف طريقة برمجة السيرفر قليلاً وذلك ان المبدء مختلف حيث انه في (أ) وضعنا زر يقوم بطلب الاتصال ثم زر آخر يرسل رسالة اما هنا نحن لن نقوم بطلب الاتصال وانما سوف نقوم بانتظار الاتصال وكذلك انتظار اي رسائل من البرنامج (أ) .. وسوف نجعله بسيطاً جداً بحيث يعرض فقط الرسائل التي تأتيه من البرنامج (أ) ... وسوف نستخدم الاحداث التابعة للاداه Winsock كثيراً هنا لذي يمكنك الرجوع الى قائمة

الاحداث السابقة للتذكر

الان ابدء مشروع جديد ثم ضع الاداه Winsock عليه وسميها wsk للاختصار

كما نعلم جميعاً فان دور السيرفر الاساسي الان هو انتظار الاتصال على المنفذ الذي سوف يتصل عليه البرنامج (أ) وهو في مثالنا المنفذ رقم ١٠٠٠٠ اذا سوف نختار الحدث Load التابع للفورم لكي نجعله ينتظر الاتصال فور تشغيل البرنامج :

```
Private Sub Form_Load()  
    wsk.LocalPort=10000  
    wsk.Listen  
End Sub
```

في السطر الاول حددنا المنفذ الذي سوف نستمع الى الاتصال منه وذلك بجعل LocalPort=10000 ثم قمنا باستدعاء امر الانتظار او الاستماع ونلاحظ هنا انه في البرنامج الاول استخدمنا الخاصية RemotePort لتحديد المنفذ بينما في السيرفر استخدمنا الخاصية LocalPort لذلك وجب الانتباه لهذا.

اذا يمكن قراءة اوامرنا السابقة كالتالي : "قم بانتظار الاتصال على المنفذ المحلي رقم ١٠٠٠٠"

نلاحظ انه عندما تقوم بالضغط على الزر اتصال في البرنامج (أ) انه قبل ان يتم الاتصال يتم اولاً اعلام السيرفر بطلب الاتصال ولن يتم الاتصال حتى يقبل السيرفر هذا الاتصال بواسطة الامر Accept كما اوضحت في جدول الاوامر السابق

اذا يتم استخدام الحدث ConnectionRequest التابع للاداه wsk حيث انه هو الذي يفيد بطلب الاتصال من العميل (البرنامج أ) ونكتب داخله اننا نوافق على الاتصال كالتالي :

```
Private Sub wsk_ConnectionRequest(ByVal requestID As Long)  
    wsk.Close  
    wsk.Accept requestID  
End Sub
```

نقوم اولاً باغلاق الاتصال السابق اذا كان موجوداً بالامر Close ثم نقبل الاتصال الجديد بالامر Accept والمتغير RequestID يحتوي على رقم الطلب لذى لا تشغل بالك فيه يكفي ان تكتب ما سبق

الان علمنا انه عندما يطلب البرنامج (أ) الاتصال عن طريق الضغط على الزر اتصال اننا سوف نقبل الاتصال وبذلك يتم الاتصال ولكن ماذا يحدث عندما يرسل البرنامج (أ) رسالة لك عن طريق الزر ارسال كما ذكرت سابقاً ؟؟

عندما يستخدم اي برنامج الامر SendData كما فعل البرنامج (أ) لارسال الرسالة فان الحدث DataArrival التابع للبرنامج الثاني ينشط للاعلام بوصول بيانات جديدة يمكن اخذها عن طريق GetData لذا بمجرد وصول الرسالة من (أ) ينشط الحدث DataArrival في (ب) ويمكنك استخدامه للحصول على البيانات او الرسائل كالتالي :

```
Private Sub wsk_DataArrival(ByVal bytesTotal As Long)  
    Dim S As String  
    wsk.GetData S  
    MsgBox S  
End Sub
```

في السطر الاول عرفنا متغير جديد من النوع String باسم S لكي نضع الرسالة فيه ثم استخدمنا الامر GetData لمعرفة الرسالة ووضعها في المتغير S ولاننا لا نريد ان نعمل بهذه الرسالة اي شيء سوى ان نعرضها قمنا باستخدام الامر MsgBox لعرض مربع حوار به هذه الرسالة

هنا فقط نكون قد انتهينا من البرنامج (ب) يمكن الان ان نشغل البرنامج (أ) ثم نشغل البرنامج (ب) بحيث يعملان سوياً وبعد ذلك ننقر على الزر اتصال في (أ) حتى يحدث الاتصال ثم اخيراً نكتب الرسالة التي نريدها في مربع النص في (أ) ونضغط على الزر ارسال فنرى الرسالة التي كتبناها قد ظهرت في البرنامج (ب) وهذا ما يحدث بين جهاز وآخر بحيث نرسل رسالة من كمبيوترك الى كمبيوتر آخر بعيد بنفس الطريقة المهم هو رقم الاي بي وكذلك ان يكون برنامج السيرفر شغال عنده بحيث اذا طلب البرنامج (أ) الاتصال يجد البرنامج (ب) ينتظر اتصاله فيقبله ويحدث بذلك الاتصال

طبعاً المثال بسيط ولم نراعي فيه تجنب الاخطاء وذلك لتوضيح الصورة مبدئياً لدى في حالة ظهور اي خطأ قم باغلاق البرنامج واعد تشغيلهما مرة اخرى وابدء الاتصال كما سبق بالنقر على اتصال ثم كتابة رسالة ثم الزر ارسال ولا تضغط على ارسال قبل اتصال اي ان الزر اتصال اضغطه مرة واحدة حتى يتم الاتصال بشرط ان يكون السيرفر شغال حينها حتى يقبل الاتصال ثم ارسل ما بدى لك من الرسائل الى (ب) باستخدام الزر ارسال....

ان اهم شيء فيما سبق ليس البرنامج وانما فقط ما يحدث بالضبط منذ بداية الاتصال وحتى قبول الاتصال وارسال الرسالة و عرضها و و و

لدى سوف اقوم بالتذكير بما يحدث بشكل سريع اولاً نحدد رقم الـ اي بي والمنفذ للجهاز الذي نريد الاتصال به عن طريق RemoteHost و RemotePort ثم نقوم بطلب الاتصال بالامر Connect ولان البرنامج الذي نريد الاتصال به (البرنامج ب) موجد على نفس الجهاز استخدمنا رقم الـ اي بي الافتراضي ١٢٧,٠,٠,١ ولكي نرسل رسالة استخدمنا الامر ... SendData هذا في كلة البرنامج (أ) ...

اما في البرنامج (ب) فقد قمنا اولاً بانتظار الاتصال وحددنا المنفذ عن طريق LocalPort ثم امر الاستماع او الانتظار Listen وبذلك نكون قد انتظرنا الاتصال ولكن عندما يطلب الطرف الآخر الاتصال ينشط الحدث ConnectionRequest لاعلامنا بذلك فنقوم بقبول او رفض هذا الاتصال ولاننا نريد الاتصال فقد قبلناه باستخدام الامر Accept وبذا يكون الاتصال قد تم ولكن بقي انه عندما يرسل الطرف الآخر بيانات او رسائل ينشط الحدث DataArrival لاعلامنا بذلك فنعمل بالبيانات ما نريد بعد اخذها بالامر GetData وهذا كل شيء حتى الان....

اتمنى ان تكون مكنيكية العمل قد فهمت بشكلها الصحيح لانه من هذه الاساسيات يمكن الانطلاق الى مدى بعيد ...

الدرس الثاني

والان دعونا اخوتي نناقش خاصية مهمة جداً في هذه الاداء وهي الخاصية State والتي نستطيع بواسطتها معرفة حالة الاتصال بين البرنامجين مثلاً اذا كنت قد عملت اتصال بين (أ) و(ب) وكان البرنامجان في اجهزة منفصلة ثم قمت بتبادل الرسائل وقام الطرف الآخر باغلاق برنامج المحادثة اثناء التشغيل في مثالنا السابق اذا قمت انت بارسال رسالة وكان البرنامج الآخر قد اغلق لاي سبب كان مثل (اعادة تشغيل الجهاز او عن غير قصد المهم انه مغلق) فانه سوف تظهر عندك رسالة خطأ ثم يتوقف برنامجك بينما في الماسنجر مثلاً فإنه يتم ابلاغك بان الرسالة لن تصل الى المستخدم الآخر لدى سوف نسعى الان لوضع معلومة ببسطة عن حالة الاتصال بحيث تستطيع من خلالها ان تميز اذا كان بإمكانك ان ترسل رسالة ام لا

طبعاً الذي يفيدنا هنا هو الخاصية State حيث ان لها قيم مفيدة لتعطي حالة الاتصال الحالي وهي كالتالي :

State = 0	مغلق
State = 1	مفتوح
State = 2	جاري انتظار الاتصال
State = 3	الاتصال معلق الان
State = 4	الطرف الآخر يعالج البيانات
State = 5	الطرف الآخر انهي المعالجة
State = 6	جاري الاتصال الان
State = 7	متصل
State = 8	قام الطرف الآخر بقطع الاتصال
State = 9	حصل خطأ ما

يتضح مما سبق انه اذا فحصنا الخاصية State وكان الناتج = ٧ فان ذلك يعني ان البرنامج متصلان اما غير ذلك فانهما غير متصلان او حالة انتظار

ويمكن استغلال هذه الخاصية الجميلة في معرفة ما اذا كان المتصل قد قام باغلاق الاتصال كما انه كتطبيق جيد ايضاً ان نجعل Label يظهر للمستخدم حالة الاتصال الحالية ...

كما ان هناك امر مهم اذا قام العميل باتصال مع السيرفر وبعد ان نجح الاتصال قام شخص باغلاق برنامج العميل فجئة ثم اعاد تشغيله بعد ذلك وضغط على الزر اتصال هل سيحدث الاتصال طبعاً لا لماذا ؟ وذلك لانه عندما تم تشغيل السيرفر كانت حالته ٢ (يعني منتظر اتصال انظر الى الجدول اعلاه) وذلك لاننا استخدمنا الامر Listen في بداية التشغيل وعندما قام العميل بطلب الاتصال كان السرفر بانتظاره فحصل الاتصال بينهما وبالتالي فان حالة البرنامج عندها اصبحت ٧ (اي متصل) اما عندما قام العميل بلاغلاق فجئة اصبحت حالة السيرفر ٨ (تنبيه باغلاق الطرف الآخر) اي انه ليس في حالة انتظار فلما حاول العميل بعد ذلك الاتصال مرة اخرى لم يكن السيرفر بانتظاره ففشلت عملية الاتصال جرب البرنامج في الدرس السابق وسوف ترى ذلك الخطأ

طبعاً الحل بسيط وذلك كله باستخدام هذه الخاصية الرائعة State حيث يمكننا مثلاً ان نضع مؤقت Timer يقوم بعرض حالة الاتصال للمستخدم بحيث اذا صادف ان الحالة اصبحت ٨ (اي قطع الاتصال) او ٩ (اي حدث خطأ ما) يقوم السيرفر بالتالي:

- 1- اظهار رسالة للمستخدم بانتهاء الاتصال ..
- 2- اغلاق الاتصال السابق الذي حصلت فيه مشكلة باستخدام الامر Close
- 3- يقوم بتحديد المنفذ من جديد عن طريق ... LocalPort
- 4- يستدعي الامر Listen بحيث يعود الى حالة انتظاره مرة اخرى كي ينتظر اتصال العميل ..

```
Select Case wsk.State
Case 0: Label1.Caption = "الاتصال مغلق"
Case 1: Label1.Caption = "الاتصال مفتوح"
Case 2: Label1.Caption = "جاري انتظار الاتصال"
Case 3: Label1.Caption = "الاتصال معلق الان"
Case 4: Label1.Caption = "الطرف الآخر يعالج البيانات"
Case 5: Label1.Caption = "الطرف الآخر انتهى معالجة البيانات"
Case 6: Label1.Caption = "جاري الاتصال الان"
Case 7: Label1.Caption = "متصل"
Case 8: Label1.Caption = "يقوم الطرف الآخر بقطع الاتصال"
Case 9: Label1.Caption = "خطأ"
End Select

' نتأكد من وجود خطأ فنقوم باعادة الانتظار
If wsk.State = 8 Or wsk.State = 9 Then
    MsgBox "الاتصال في خطأ"
    wsk.Close
    wsk.LocalPort = 10000
    wsk.Listen
End If
```

لاحظ اننا استخدمنا العبارة Select Case وذلك لاطهار حالة الاتصال للمستخدم بحيث يكون على دراية تامة بما يحصل الان في عملية الاتصال

طبعاً كل ما سبق يعطي برنامجنا شيئاً من القوة ولكن بقي امر لم نقوم به فماذا اذا قام احد البرنامجين بالضغط على الزر ارسال قبل ان يكون الاتصال قد تم فعلاً اي قبل الضغط على الزر اتصال في العميل النتيجة طبعاً هي خطأ فيمكن ان نستخدم SendData لارسال البيانات قبل ان يكون هناك اتصال فعلاً وما العمل في هذه الحالة ؟؟؟؟

طبعاً العمل بسيط وهو التاكيد بان هناك اتصال قبل ارسال البيانات عن طريق ... State حيث كانت الشفرة في الزر ارسال كما يلي:

```
Wsk.SendData "" : رسالة من أ  
DoEvents
```

ويجب ان تصيح هكذا :

```
If wsk.State = 7 Then  
    wsk.SendData "" : رسالة من أ  
    DoEvents  
End If
```

وذلك في كلاً من العميل والسيرفر بحيث يتم التحقق من هناك اتصال فعلاً ثم ارسال البيانات بعد ذلك .

بقي نقطة اضافية في الدرس السابق حيث كنا عندما يقوم المرء باضغط على الزر اتصال نقوم باغلاق الاتصال السابق ثم نطلب اتصال جديد وذلك كالتالي :

```
wsk.Close  
wsk.RemoteHost = "127.0.0.1"  
wsk.RemotePort = 10000  
wsk.Connect
```

كان ذلك قبل ان نتعرف على الخاصية State اما الان فما هو الداعي لاغلاق الاتصال بالامر Close اذا كان الاتصال يعمل بشكل سليم !!!! لذي سوف نتأكد أولاً اذا كانت الحالة هي 7 اي الاتصال قائم اذا لا داعي لفعل شيء اما غير ذلك فانه يوجد خطأ او ماشابه كأن يقوم الخادم بالاغلاق فيجب عندها ان نغلق الاتصال الحالي ثم نطلب اتصال جديد ... كالتالي :

code:

```
If wsk.State <> 7 Then  
    wsk.Close  
    wsk.RemoteHost = "127.0.0.1"  
    wsk.RemotePort = 10000  
    wsk.Connect  
End If
```

(العلامة <> تعني لايساوي)

وبذلك نجدد الاتصال اذا كان الاتصال الحالي به قصور فقط ...

اخوتي ان المفاهيم السابقة على الرغم من بساطتها فهي مهمة جداً في التخلص من رسائل الخطأ التي قد تنتج دائماً من استخدام Winsock حيث يجب ان نكون متأكدين من وجود اتصال قبل اي عملية لتبادل البيانات كما انه يجب ضمان ان يعود الخادم (السيرفر) الى حالة انتظار اذا ما حصل اي خطأ في الاتصال وهو ما قمنا به سوياً في هذا الدرس لذي يجب على الجميع اتقانه جيداً قبل الاستمرار في الدروس القادمة...

الدرس الثالث

سوف نبني في هذا الدرس اول برنامج للمحادثة بين شخصين يشبه الماسنجر في العمل العام ... حيث تظهر الرسائل التي تتلقاها من الطرف الآخر على شكل قائمة بها اسمه ثم الكلام الذي بعثه

برنامج العميل :

نقوم باضافة ثلاثة مربعات نص للبرنامج السابق ونسميهم :

txtName

وهو المربع الذي سوف يحمل اسمك في الحوار


```
txtIP
المربع الذي سيحتوي على رقم الاي بي الذي نريد الاتصال به
txtList
مربع يحتوي على الرسائل المرسله والمستقبله
```

ونجعل الخاصية MultiLine=True لمربع النص txtList بحيث يسمح بادخال اكثر من سطر بحيث تبدو على شكل قائمة يوضع فيها الحوار الذي يدور بين العميل والخادم ...
والذي سوف يتغير هو انه عندما تصلنا رسالة من الخادم فاننا نريد اضافتها الى اسفل القائمة وليس عرضها بمربع حوار لذى نستبدل الكود في الحدث DataArrival كالتالي :

```
Dim S As String

wsk.GetData S

txtList = txtList & vbNewLine + vbNewLine + S

txtList.SelStart = Len(txtList)
```

بعد قراءة البيانات المرسله وجعلها داخل المتغير S نقوم بوضعها في اسفل القائمة وذلك بعد سطرين فارغين vbNewLine والذي يمثل سطر جديد وفي الاخير نريد ان نسحب التركيز الى اسفل القائمة وذلك بالسطر i=txtList.SelStart = Len(txtList) وبدونه سوف تمتلىء القائمة ثم يغيب النص الجديد في الاسفل دون ان نراه كما هو ملاحظ في الماسنجر عندما يقوم الشخص المتحدث بارسال رسائل لك فان شريط التمرير ينزل للأسفل بحيث يعرض لك آخر ما تم ارسالة من نص وهذا ما يحدث بالضبط بواسطة هذا السطر

كما انه عندما يضغط احد على الزر ارسال يتم ارسال النص المكتوب في المربع txtMsg ويكون الاسم المكتوب في المربع txtName امامه بحيث يرى الاسم الذي وضعته لنفسك امام النص وذلك كله في الزر ارسال كالتالي :

```
If wsk.State = 7 Then
wsk.SendData txtName + " : " + txtMsg
DoEvents
txtList = txtList & vbNewLine & vbNewLine & txtName & " : " & txtMsg
txtMsg = ""
End If
```

بقي شيئاً واحداً هو انه عند الضغط على Enter يجب ان يتم ارسال النص وكانك نقرت على الزر ارسال وذلك يكون بفحص الزر المضغوط عليه في الحدث KeyPress التابع لمربع النص فاذا كان الزر هو Enter يتم النقر على الزر ارسال لارسال البيانات .. وذلك كما يلي :

```
Private Sub txtMsg_KeyPress(KeyAscii As Integer)
If KeyAscii = 13 Then Command1 = True
End Sub
```

حيث ان الرقم ١٣ هو رقم الزر Enter وبذلك نكون قد انهينا برنامج العميل

حيث ان كل زر على لوحة المفاتيح له رقم خاص و لكنك غير ملزم بحفظ ارقام جميع المفاتيح حيث يوفر لك الفيجوال بيسك اختصارات لتلك المفاتيح مثلاً بدل من ان نكتب الرقم ١٣ نكتب vbKeyReturn واذا اردنا المفتاح D مثلاً نكتب vbKeyD وهكذا ...

اما الخادم فهو بنفس التعديل الذي قمنا به في العميل مع فرق بسيط حيث انه لاداعي لاضافة مربع لوضع رقم الاي بي لانه لن يتم الاتصال من الخادم فقط يقوم بالانتظار

الدرس الرابع

اما الان دعونا نفكر بشي من التفصيل نحن دائماً نقوم بارسال رسائل نصية ولكن كيف يمكننا ان اؤدي مهام غير عرض الرسائل او لنقل مبدئياً اذا كان لدينا برنامج للعميل وفيه مربع نص txtMsg واربعة ازرار واحد هو للاتصال ... اما الثلاثة الآخرون فاسمائهم :

- 1 - Text
- 2 - Message
- 3 - Caption

واريد انه اذا نقر على الزر الذي اسمه Text يتم عرض الرسالة التي في مربع النص txtMsg في مربع نص اسمه Text1 داخل الخادم

واذا نقر المستخدم على الزر Message فان الرسالة تعرض بواسطة مربع حوار في برنامج الخادم

وكذلك عند النقر على الزر Caption في العميل يتم عرض الرسالة في شريط عنوان الخادم اي عنوان الفورم التابعة لبرنامج الخادم ...

طبعاً هنا يختلف الوضع فنحن لانريد من العميل ان يخبرنا ماهي الرسالة فقط كما كان سابقاً حيث نضعها في المتغير S ثم نعرضها ولكننا الان نريد من العميل ان يخبرنا كيف يريد ان يتم عرض هذه الرسالة ايضاً وبذلك نحن نحتاج الى نوعين من المعلومات وهي نوع العرض و نص الرسالة

اي ان الطريقة السابقة لاتنفع وهي مثلاً :

```
Wsk.SendData txtMsg  
DoEvents
```

فهذه الطريقة ترسل النص الموجود في مربع النص txtMsg ولكنها لا تخبر الخادم كيف يعرض هذا النص وهنا نبدء في تغيير طريقة الارسال وطريقة الاستلام للبيانات في كلاً من الخادم والعميل..

والفكرة هي ان نضع عدد معين من الحروف ولنقل اربعة قبل كل رسالة تبين نوع العرض مثلاً :

Text	تعني قم بوضعها في مربع نص
Cptn	تعني قم بوضعها في عنوان الفورم
MgBx	تعني قم بوضعها في مربع حوار

مثلاً سيكون شكل الكود في الزر Message في برنامج العميل بالشكل التالي :

```
Wsk.SendData "MgBx" + txtMsg
```

وبذلك عندما يقوم الخادم بإستلام الرسالة يفحص الحروف الاربع الاولى ويقوم بعرض باقي الرسالة بالطريقة التي تخبره به الحروف الاربعة الاولى ...

ويكون ذلك بفصل الرسالة الى جزئين الاول وهو الامر ووضعت في متغير اسمه Cmd والباقي هو نص الرسالة ووضعت في متغير اسمه Msg ثم نقوم بفحص الامر Cmd ونحدد طريقة العرض باستخدام Select Case كالتالي :

```
Dim S As String
Dim Msg As String
Dim Cmd As String

wsk.GetData S

Cmd = Left(S, 4)
Msg = Mid(S, 5)

Select Case Cmd
Case "Text": Text1.Text = Msg
Case "MsgBox": MsgBox Msg
Case "Caption": Caption = Msg
End Select
```

الامر Left يقوم باقتطاع عدد من الاحرف حسب الرقم الذي وضعناه من يسار المتغير الذي حددنا مثلاً في مثالنا هذا يقوم الامر left بأخذ أول ٤ احرف من المتغير S وتخزينها في المتغير Cmd و الأمر Mid يقوم بأخذ عدد من الحروف بمقدار بداية و نهاية نحدده نحن و اذا لم نحدد مقدار النهاية يقوم بأخذ الحروف من مقدار البداية الى نهاية المتغير مثلاً في مثالنا قمنا بأخذ الاحرف من الحرف رقم ٥ الى نهاية المتغير S و تخزينه في المتغير Msg

الدرس على الرغم من سهولته فهو بنظري من اهم المواضيع لفهم تنظيم تبادل البيانات والاوامر بين العميل الى الخادم...

حيث اننا فيما سبق كنا نستخدم مصطلح الخادم مجازياً ولقد سمي الخادم لانه يجب ان يوفر للعميل الخدمات التي يطلبها مثلاً اذا اراد العميل يعرف كم الوقت في الجهاز الذي به برنامج الخادم يقوم بارسال رسالة تبء بحروف متفق عليها بينهما مثلاً Time وعندما تصل الرسالة الى الخادم يقوم بفحص الحروف الاولى ويعرف ماذا اراد العميل بالضبط ثم يقوم بمعرفة الوقت بواسطة اوامر الوقت ثم يستخدم الامر SendData لارسال البيانات التي طلبها العميل منه

على العموم سوف نتطرق الى هكذا امور ان شاء الله مستقبلاً ونكتفي الان بمثال على عرض الرسالة باحدى الطرق الثلاث السابقة

أساسيات من اجل الدرس الخامس

رأيت انني سوف احتاج الى بعضاً من الاساسيات في المرحلة القادمة واخشى ان لا اجد لها عند البعض لدى سوف اوضح هنا امرين سوف احتاجهما في برنامج المحادثة الجماعية القادم باذن الله تعالى .. وارجوا ممن ليس لديه فكرة كبيرة فيهما ان يقرأ الشرح ثم يطبق كل شيء بحيث لا تتصعب عليه الامور بعد ذلك وهي امور بسيطة جداً فقط تحتاج الى تطبيق ...

اما الموضوع الاول فهو مصفوفة الادوات واستخدامها او Control Array والثاني هو صندوق القائمة او الـ ListBox وطريقة التعامل معها فهي مهمة جداً في الموضوع القادم ولا اريد ان اخلط بين شرح فكرة البرنامج وشرح التعامل مع هاتان الخاصيتان لدى سوف اقدم شرح سريع عنهما اتمنى ان يؤدي الغرض

اولاً : مصفوفة الادوات والتي تعني وجود ادوات فوق الفورم من نفس النوع وب نفس الاسم يفرق بينهم بالرقم مثلاً ان يكون عندنا ثلاثة ازرار كلها باسم Command1 ولكي نفرق بينها نفرق بينها بالرقم او الـ Index مثلاً :

```
Command1(0).BackColor = vbRed
Command1(1).Caption = "Welcome"
Command1(2).Width = 800
```

في الكود السابق الزر الاول رقمة ٠ وقمنا بتغيير لونه الى الاحمر والزر الثاني رقمه ١ وقمنا بجعل عنوانه هو Welcome والزر الاخير او الثالث رقمه ٢ وجعلنا عرضه ٨٠٠

واذا نظرنا الى الحدث Click لاي زر فانه قبل ان يكون هناك ازرار بنفس الاسم فهو على الشكل التالي:

```
Private Sub Command1_Click()

End Sub
```

وبعد ان نضيف الى الفورم عناصر اخرى بنفس الاسم فانه يصبح :

```
Private Sub Command1_Click(Index As Integer)

End Sub
```

والملاحظ هنا انه تم اضافة متغير جديد اسمه Index وهو الذي يحدد الزر الذي تم النقر عليه ... وبذلك فاذا اضعنا مثلاً ثلاثة ازرار بنفس الاسم مع تغيير الخاصية Index اي ان لكل منهم رقماً فانه لن يكون لكل زر من الازرار حدث Click خاص به ولكن كل الاحداث تكون مشتركة ويمكنك تمييز الزر الذي تم النقر عليه بفحص القيمة Index لنفترض اننا نريد ان نجعل عنوان الفورم One اذا تم النقر على الزر رقم واحد و Two اذا نقر على الزر رقم اثنان و Three في حالة النقر على الزر الثالث فاننا سوف نكتب التالي:

```
Private Sub Command1_Click(Index As Integer)

    Select Case Index
        Case 0: Caption = "One"
        Case 1: Caption = "Two"
        Case 2: Caption = "Three"
    End Select

End Sub
```

وذلك لان الترقيم يبدأ من صفر.... وكل ما سبق كان لاننا سوف نحتاج الى اضافة اكثر من اداة Winsock باسم واحد ولنقل مثلاً wsk بالتالي فاننا سوف نميز بينها بالرقم كما فعلنا مع الازرار السابقة الذكر

ثانياً : مربع القائمة وهو ListBox يستخدم لعرض بيانات او عناصر في قائمة بحيث يمكنك تحديدها واختيار احداها وعندما تريد ان تضيف عنصر الى القائمة ولنفرض ان اسم القائمة هو List1 فاننا سوف نستخدم الامر AddItem للاضافة لها بالشكل التالي:

```
List1.AddNew "كلنا عبيد لله سبحانه"
```

و الكود السابق سوف يضيف العبارة "كلنا عبيد لله سبحانه" الى مربع القائمة واذا اردنا اضافة المزيد نستخدم نفس الكود...

وكذلك الامر RemoveItem يستخدم لحذف عنصر من القائمة ويجب ان تمرر له رقم العنصر الذي تريد حذفه هل الاول او الثاني او بحيث يبدأ الترقيم من الصفر اي انك لكي تحذف العنصر الاول في القائمة اكتب التالي :

```
List1.RemoveItem 0
```

وصفر يعني العنصر الاول بينما ١ يعني العنصر الثاني وهكذا ...

واما الخصائص فهي كما يلي :

```
Text
هي عبار عن نص يحتوي على العنصر المحدد من القائمة
List
عبارة عن مصفوفة بالعناصر كاملة
ListCount
هو عدد العناصر في القائمة
ListIndex
يحمل رقم العنصر المحدد ويكون (-١) اذا لم يكن هناك عنصر محدد
```

وبالتالي لنفترض اننا نريد ان نطبع كل عناصر القائمة في الفورم فاننا سوف نستخدم التالي :

```
Dim I As Integer

For I = 0 To List1.ListCount - 1
    Print List1.List(I)
Next
```

حيث اننا عملنا دوران يبدأ من الرقم صفر والذي يعني اول عنصر في المصفوفو الى النصر الذي رقمه هو List1.ListCount-1 وهو العنصر الاخير في القائمة لان الترقيم يبدأ من الصفر وبالتالي اذا كان هناك ثلاثة عناصر في القائمة اي ان ListCount=3 فان رقم العنصر الاخير هو ٢ وهو.... (ListCount-1)

يجب على الجميع فهم ما سبق جيداً كي اتمكن من ان اشرح الموضوع التالي بدون الالتفات الى هذه المبادئ الاساسية

الدرس الخامس

اخوتي سنبدأ هنا بتوضيح الفكرة العامة للمحادثة الجماعية بشكل سريع ثم نلحقه بعد ذلك بمثال وشرح لهذا المثال حتى نستفيد جميعاً من الناحية النظرية والتطبيقية...

في السابق كنا نستخدم اسم الخادم مجازياً فلم يكون يقوم بدور آخر سوى الانتظار وعندما يتم الاتصال يصبح الخادم له نفس دور العميل في الارسال والاستقبال وفي الاصل الخادم لم يسمى بهذا الاسم الا لانه يقوم بتقديم خدمات للعميل اي انه يوفر له المعلومات والبيانات التي يريدتها العميل كما ينظم سير عملية التواصل بين الاطراف اما في برامج المحادثة السابقة لم نرى الدور الفعلي له

اما في موضوع المحادثة الجماعية فان كل مستخدم مشترك في المحادثة سيكون لديه عميل مستقل ولن يكون شرطاً على كل عميل بالاتصال بكل عميل آخر حتى يستطيع تبادل الرسائل بينهما وانما سوف يتصل الجميع بالخادم والذي سيكون بدوره متصلاً بالجميع وعندما يقوم احد الاشخاص بارسال رسالة من برنامج العميل الذي بحوزته الى الخادم فان الخادم سيقوم باستلام الرسالة ثم يقوم بدوره بارسالها الى جميع الاعضاء المشتركين في المحادثة وبذلك فان عملية تنظيم المحادثة بين الاطراف سوف تقع على عاتق الخادم الذي سيضمن تواصل الجميع بعضهم ببعض ... ولكن كيف يمكن للخادم القيام بهذا الدور وكذلك ان يكون متصلاً باكثر من جهة اتصال (باكثر من عميل واحد) وهذا ما سوف نناقشه معاً الان

سوف تختلف الفكرة الان من حيث ان الخادم لن يرتبط بعميل واحد بل يجب ان يمتلك القدرة على الارتباط باكثر من عميل بحيث يبقى منتظراً للاتصال من اكثر من جهة ولانه ليس للداه Winsock القدرة على الارتباط باكثر من جهة في نفس الوقت لذي تظهر الحاجة الى ان يمتلك الخادم اكثر من اداه winsock بحيث نقوم بتحقيق الاتصال بكل عميل يحاول الاتصال به (وهنا سوف نحتاج الى مصفوفة من ادوات winsock.

قد يتبادر الى الذهن مما سبق اننا سوف نجعل اكثر من اداه في الخادم بحيث تنتظروا هذه الادوات الاتصال على نفس المنفذ طبعاً لا لانه لايمكن ان تستخدم اكثر من جهة المنفذ ذاته في نفس الوقت فعندما نستدعي الامر Listen نقوم الاداه بحجز المنفذ المحدد بالخاصية LocalPort ولايمكن ان يتم استغلاله بواسطة اداه اخرى الا في حالة ان الاداه غيرت حالتها (State=2 ويعني انتظار) الى اي شيء آخر سوى ان تصبح متصلة او تنهي الانتظار بواسطة Close وعندما يتحقق الاتصال فانه عندما نقبله بواسطة Accept ينتهي الانتظار ويصبح من الممكن استغلال المنفذ مرة اخرى وليس من المعقول ان نجعل كل اداه تحجز منفذ مختلف فهذه طريقة مكلفة ومربكة حيث انك سوف تحجز عدد لا بأس به من منافذ النظام كما ان العميل عندما سيحاول الاتصال يجب عليه تجربة الاتصال بكل المنافذ المفتوحة فلربما انه طلب الاتصال بالمنفذ ١٠٠٠٠ مثلاً ووجده مرتبط مع شخص آخر يجب عليه بعدها ان يجرب المنفذ ١٠٠٠١ حتى يصل الى منفذ فارغ ويتصل به وهذا حل غير عملي

بل ان كل ما سبق كان مجرد حلول ضعيفة ذكرتها حتى نتعايش معاً مع المشكلة ونتعود دائماً على تقليب كل الحلول في رؤوسنا للوصول الى الحل الامثل

اما الحل المناسب في رأيي هو ان نقوم بوضع اكثر من اداه winsock في الخادم ولنقل ١١ اداه بحيث نجعل الاداه الاولى اي رقم 0 نقوم بانتظار الاتصالات الخارجية على منفذ محدد يعرفه كل العملاء ويقومون بالاتصال به وقد كنا في السابق نقوم بقبول الاتصال بمجرد وصول طلب الاتصال الينا من العميل كالتالي :

```
Private Sub wsk_ConnectionRequest(ByVal requestID As Long)

    wsk.Close
    wsk.Accept requestID

End Sub
```

اما الان اذا فعلنا ذلك فان الاداه سوف تنهي حالة الانتظار ولن يكون بمقدورها استلام اي اتصال من اي عميل آخر لذي فنحن هنا لن نعمل ذلك ولكن الاداه المؤلة عن الانتظار في حالتنا الجديدة هذه لن تقوم بقبول الاتصال مباشرة بل سوف تبحث في الادوات العشر المتبقية بحيث اذا صادفت اداه فارغة اي

غير متصلة انها مغلقة وحالتها State=0 وتجعلها تقبل الاتصال الجديد وبالتالي تبقى هي (الاداه رقم ٠) على حالتها السابقة في انتظار اي اتصال جديد وذلك على النحو التالي :

```
Private Sub wsk_ConnectionRequest(Index As Integer,
ByVal requestID As Long)
    Dim I As Integer

    For I = 1 To 10
        If wsk(I).State = 0 Then
            wsk(I).Accept requestID
            Exit For
        End If
    Next

End Sub
```

لا حظ اننا بدأنا البحث من الاداه رقم واحد متجنبين بذلك الاداه رقم ٠ كي تبقى هي حلقة الوصل بين برامج العملاء والخادم...

اما البيانات التي سوف يتلقاها الخادم من العميل هي عبارة عن احدى رسالتين الاولى تبدأ ب NewMssg كا امر متبوع بنص الرسالة وهو عبارة عن رسالة يبعثها العميل ولان كل عميل هو متصل بالخادم وليس بعميل آخر فانه لن يبعث الرسائل الا الى الخادم وتقع على الخادم مهمة ارسالها الى باقي العملاء بحيث تظهر عندهم بانها من العميل فلان ... حيث يقوم الخادم عند استلامه هذا الامر بالبحث عن جميع العملاء المتصلين اي التي تكون حالة الاداه المرتبطة بهم هي State=7 ويقوم بارسالها اليهم بحيث يقوموا بعرضها في مربعات الرسائل لديهم ...

هذا كان بالنسبة للرسالة الاولى اما الرسالة الاخرى التي يمكن ان يتلقاها الخادم من العميل هي الرسالة NewUser ثم يليها اسم المستخدم فعندما يتم الاتصال بين العميل والخادم يجب على العميل ارسال هذه الرسالة قبل اي شئ آخر بحيث يليها اسم العميل حتى يقوم الخادم بارسال رسالة الى كل العملاء الآخرين بان العميل المسمى فلان قد اضيف الى المحادثة وذلك في برنامج الخادم كالتالي :

```
Private Sub wsk_DataArrival(Index As Integer, ByVal
bytesTotal As Long)
    Dim Cmd As String
    Dim Msg As String
    Dim S As String
    Dim I As Integer

    wsk(Index).GetData S

    Cmd = Left(S, 7)
    Msg = Mid(S, 8)

    ' -----
    If Cmd = "NewMssg" Then

        For I = 1 To 10
            If wsk(I).State = 7 Then
                wsk(I).SendData "Message" + Msg
                DoEvents
            End If
        Next

    End If

    ' -----
    If Cmd = "NewUser" Then
```

```

        Num = Num + 1

        For I = 1 To 10
            If wsk(I).State = 7 Then
                wsk(I).SendData "MssgBox" + Msg
                DoEvents
            End If
        Next

    End If

End Sub

```

لاحظ ان طول الامر Cmd هنا هو ٧ حروف ... كما يجدر الاشارة بان الخادم عند استلام هذه الرسائل من العميل يقوم بارسال احدى رسالتين الى كل العملاء الاخرين وهذه الرسالتان هما اما MssgBox او Message يستلم الخادم رسالة NewMsg من احد العملاء يرسل الرسالة Message الى كل العملاء الاخرون الذي يقومو بعرض محتويات هذه الرسالة في قائمة المحادثة ... اما اذا استلم الخادم الرسالة NewUser وذلك عند تسجيل عميل جديد فانه يرسل الرسالة MssgBox الى باقي العملاء كي يقومو بعرض مربع حوار يفيد بان مستخدم جديد قد اضيف الى المحادثة....

وذلك كالتالي في العميل كالتالي :

```

Private Sub wsk_DataArrival(ByVal BytesTotal As Long)
    Dim S As String
    Dim Cmd As String
    Dim Msg As String

    wsk.GetData S
    Cmd = Left(S, 7)
    Msg = Mid(S, 8)

    Select Case Cmd
        Case "Message": txtList = txtList & vbNewLine +
vbNewLine + Msg
        Case "MssgBox": MsgBox " لقد تمت اضافة المستخدم " +
Msg + " الى المحادثة "
    End Select

    txtList.SelStart = Len(txtList)
End Sub

```

نلاحظ انه يوجد فقط نوعين من الرسائل التي يتم تبادلها بين العميل والخادم وهذه طبعاً لا تكفي للقيام بمعظم محاور برامج المحادثة ولناخذ مثلاً ان المستخدمين لا يعلمون كم عدد الافراد الذين يتحدثون معهم الان كما انه اذا غادر احدهم المحادثة فلا يعلم الباقيون بمغادرته وذلك اننا لم نضمن ذلك في هذا البرنامج الذي تعمدت جعله بسيطاً حتى اناكد من تسرب الفكرة الى رؤوسنا ثم نناقش بعد ذلك اضافات على الموضوع لذي ارجو ان يقوم الجميع بفهم الدرس جيداً لان الدرس الذي يليه سيكون معقداً بشكل اكبر فمن اراد الاستمرار بشكل سليم فعليه فهم كل سطر في البرنامج والمهم هو فهم الفكرة العامة بحيث يمكن تطويرها بعد ذلك لتناسب احياجاته مع اضافة مميزات اكثر عليها ...

الدرس السادس

اخوتي الكرام لقد رأينا سابقاً كيف يمكن ان نكون برنامج محادثة لاكثر من شخص ولكن كان ذلك بمثابة البداية حيث لم يحتوي على اشياء مطلوبة بشده وهي ظهور قائمة بالموجودين حالياً كما انه يجب ان يعرض تنبيهاً اذا غادر اي شخص المحادثة لذى نقوم هنا بتعديل البرنامج السابق كي يتضمن هذه الميزة المهمة

سوف نضيف ListBox الى كلاً من الخادم والعميل وذلك بغرض وضع الاسماء داخل هذه القائمة ثم نجعل العميل عند اتصاله بالخادم مباشرة يرسل رسالة فيها الامر NewUser ثم اسم المستخدم الجديد وعندما يستلم الخادم هذه الرسالة يقوم مباشرة باضافة العميل الى قائمة المستخدمين عنده ثم يرسل رسالة الى جميع العملاء الآخرين عنوانها NewUser وبها اسم العميل الجديد ثم يرسل ايضاً رسالة اخرى هي NewList وبها اسماء الاعضاء جميعاً بحيث يقوم كل عميل عندما يستلم هذه الرسالة بمسح القائمة التي لديه ثم يضيف اليها الاسماء التي جاءت في رسالة الخادم الامر الى هنا يبدو سهلاً ولكن كيف يمكن ان ارسل محتويات القائمة ونحن نعلم ان القائمة بها اكثر من اسم ونحن نريد ان نرسل رسالة بها كل اسماء المستخدمين بحيث لا تختلط هذه الاسماء مع بعضها البعض ويتمكن العميل من استخراج جميع الاسماء واصافتهم الى القائمة الخاصة به طبعاً نحن لن نقوم بارسال الاسماء في القائمة واحداً بعد الآخر ولكننا سوف نرسل جميع الاسماء دفعة واحدة بحيث نقوم بالفصل بينهم بفاصل معين ولنقل مثلاً الرمز "#" ... وبذلك اذا كان لدينا قائمة بها اربعة مستخدمين هم علي و وليد وهيثم و ابو عابد فان الخادم يقوم بجعل الرسالة كالتالي : "علي#وليد#هيثم#ابو عابد" ويضيف لها الامر NewList بحيث يعلم العميل بانه يجب عليه ان يضعهم في قائمة جديدة بعد ان يفصلهم عن بعضهم البعض ... ولكي نقوم بعملية تكوين هذا الشكل من قائمة المستخدمين في الخادم نستخدم كود كالتالي وقد اسميته هنا: GetList

```
Private Function GetList() As String
    Dim S As String
    Dim I As Integer

    S = ""
    For I = 0 To lstUsers.ListCount - 1
        S = S + "#" + lstUsers.List(I)
    Next

    GetList = Mid(S, 2)
End Function
```

حيث قمنا بعمل دوران لكل العناصر في القائمة lstUsers والتي تحتوي على كل المستخدمين في الخادم ووضعنا قبل كل اسم الرمز... "#" طبعاً عندما تصل هذه الرسالة الى العميل وهي تحمل الامر NewList سوف يكون على العميل ان يقوم بفصل جميع الاسماء عن بعضها البعض ثم يضعها في القائمة عنده ولنفترض ان اسم القائمة في العميل هي lstUsers ايضاً .. سوف نحتاج لفصل الاسماء الى الامر Split وهو يقوم بفصل اي نص يحتوي على فاصل معين وهو في مثالنا "#" ثم يضع الناتج في مصفوفة وبذلك يمكننا ان نضيف عناصر هذه المصفوفة الى القائمة lstUsers كالتالي وقد سميته هنا: FillList

```
Private Sub FillList(S As String)
    Dim A() As String
    Dim I As Integer

    On Error Resume Next
    A = Split(S, "#")

    lstUsers.Clear
    For I = LBound(A) To UBound(A)
        lstUsers.AddItem A(I)
    Next
End Sub
```

حيث ان الاسماء تكون موضوعه ضمن المتغير S ثم نقوم باستخدام الامر Split لفصل الاسماء ووضعها

في مصفوفة من النوع String واسمها A ثم بعد ذلك نقوم باضافة عناصر هذه المصفوفة A الى القائمة lstUsers اويزا نكون قد استلمنا القائمة الجديدة كاملة

هذا بالنسبة لارسال واستلام الاسماء ولكن ماذا عندما يقرر احد العملاء ان يغادر فانه يجب ان نقوم بحذفه من القائمة ونحن نعرف انه لكي نحذف عنصر من اي قائمة نستخدم الامر RemoveItem كما سبق ان اوضحنا في السابق وهو يحتاج الى رقم العنصر الذي نريد حذفه ولكننا ليس لدينا سوى اسم العنصر وليس رقمه لدى يجب ان نحصل على رقم العنصر في القائمة بواسطة اسمه الذي معنا وسوف نستخدم لذلك عملية بحث داخل القائمة عن اسم المستخدم ثم عندما نعثر عليه نأخذ رقمه ونحذفه وقد جعلت لذلك امر باسم RemoveUser ويتم اعطائه اسم المستخدم وهو يتكفل بالبحث عنه ثم حذفه من القائمة وذلك كالتالي :

```
Private Sub RemoveUser(User As String)
    Dim I As Integer

    For I = 0 To lstUsers.ListCount - 1
        If lstUsers.List(I) = User Then
            lstUsers.RemoveItem I
        End If
    Next
End Sub
```

طبعاً يستخدم هذا الكود العميل عندما تصله الرسالة DelUser متبوعه باسم المستخدم والتي تعني ان المستخدم فلان قد غادر المحادثه فيقوم باستخدام الاجراء السابق لحذفه من القائمة كما يقوم بعرض تنبيه للعميل بمغادرته للمحادثة

من الملاحظ انه في كلا مما سبق ان المسألة تعتمد بشكل كبير على التعامل مع القائمة من حذف واطافة وبحث لدى يجب علينا تفحص الكود السابق جيداً بحيث نستوعبه كاملاً لان التمارين القادمة التي سوف اضعها بعد ان تناقش اي غموض في الموضوع ان شاء الله تعالى ستكون صعبه شوي لدى يجب الاستعداد لها جيداً ...

التعامل مع الملفات

بسم الله الرحمن الرحيم ...

في الواقع لا نستطيع الخوض في ارسال الملفات دون معرفة التعامل مع الملفات بواسطة الفيجوال بيسك حيث انه يجب علينا ان تمكن من فتح وانشاء وحفظ ملف بواسطة الكود ثم بعد ذلك نتطرق الى عملية الارسال التي لن تشكل مشكله بعد ذلك ...

ويجب ان نعلم انه توجد طرق محددة للتعامل مع الملفات في الفيجوال بيسك سوف نستخدم نحن الطريقة Binary وتتميز هذه الطريقة عن الطرق الاخرى في كونها تتعامل مع الملفات كبيانات اولية وليس كنوع معين مثل Input و Output اللتان تتعاملان مع ملفات النصوص او Random والتي تتعامل مع انواع ينشئها المبرمج اما ال Binary فهي تتعامل مع كل الانواع ولدى فهي ممتازة لنقل اي نوع من الملفات سواء كان ملف صوت او صورة او قاعدة بيانات او غير ذلك وبغض النظر عن نوعه...

فلكي نفتح ملف نستخدم الصيغة التالية :

```
Dim F As Integer  
  
F = FreeFile  
Open FileName For Binary As F  
  
Close F
```

لاحظ اننا في السطر الاول عرفنا متغير F من النوع Integer حتى نخزن فيه رقم الملف المفتوح وبذلك يمكن بعد ذلك ان نستخدم رقم الملف بدلاً من اسمه مثلاً اذا اردنا ان نغلق الملف نكتب Close F وتعني اغلق لنا الملف الذي رقمه مخزن في المتغير F والدالة FreeFile تقوم بالبحث عن رقم ملف غير مستعمل اي كاننا نقول ابحت عن رقم جديد وضعه في المتغير F لكي نستخدمه مع الملف ثم بعد ذلك نستخدم الامر Open لفتح الملف الذي اسمه FileName من النوع Binary ونتعامل معه كانه الرقم F وفي الاخير قمنا بإغلاق الملف F بواسطة Close F

طبعاً المثال في السابق لم نقوم بعمل شيء سوى اننا فتحنا الملف FileName ثم قمنا باغلاقه بعد ذلك ... (لاحظ انه اذا كان الملف الذي اسمه FileName غير موجود في القرص الصلب يقوم الفيجوال بإنشاء ملف جديد بهذا الاسم)...

ولكي نستطيع ان نكتب الى هذا الملف نستخدم الامر Put ولكي نقرأ منه نستخدم الامر Get مثلاً لكي نكتب الى الملف العبارة "على الله توكلنا" نستخدم التالي :

```
Dim F As Integer  
Dim S As String  
  
S = "على الله توكلنا"  
F = FreeFile  
  
Open FileName For Binary As F  
Put #F, 1, S  
Close F
```

لاحظ اننا وضعنا العبارة في متغير نصي ثم فتحنا الملف ثم استخدمنا الامر put لادخال النص في الملف

الذي رقمه F وبداية الكتابة في الموقع رقم ١ ثم أخيراً نضع محتويات المتغير S ... حيث انه لكي نستخدم الامر Put نحتاج الى رقم الملف ثم الى الموقع الذي نريد الكتابة فيه ثم الشيء الذي نريد كتابته لاحظ انك اذا فتحت الملف مرة اخرى ثم استخدمت عبارة اخرى ووضعتها في الموقع رقم ١ فانه سوف يتم الكتابة فوق العبارة السابقة لذى لكي نضيف العبارة الجديدة الى الملف بعد العبارة السابقة يمكن ان نحدد موقع الكتابة في مكان آخر بعد العبارة السابقة ولكي نعرف طول الملف نستخدم الامر LOF والذي يرمز لـ Length Of File اي طول الملف ثم نضيف النص الجديد تخيل الان اننا نريد ان نضيف العبارة "ولا حولة ولا قوة الا بالله" نفتح الملف ثم نكتب بعد العبارة في الموقع الذي يلي العبارة السابقة كالتالي :

```
Dim F As Integer
Dim S As String

S = "ولا حولة ولا قوة الا بالله"
F = FreeFile

Open FileName For Binary As F
Put #F, (LOF(F) + 1), S
Close F
```

لاحظ التغيير الحاصل في الموقع حيث استخدمنا الامر LOF لمعلاقة طول الملف F وطوله هو ١٦ لان العبارة السابقة التي يحويها الملف طولها ١٦ حرف ثم بعد ان وجدنا طول الملف قمنا باضافة رقم واحد وبالتالي فان الكتابة سوف تتم في الموقع الذي يلي العبارة السابقة ببايت واحد وبذلك نضمن اضافة العبارة الجديدة الى العبارة السابقة وسوف نستخدم نفس المبدء في عملية نقل الملفات اذ اننا سوف نقوم بنقل البيانات على دفعات ونقوم باضافة البيانات الى الملف دفعة بعد الاخرى كما فعلنا في المثال السابق لذى يجب استيعابه جيداً....

الان ما الذي يجب علينا عمله اذا اردنا القراءة من الملف مثلاً اذا اردنا قراءة ١٠ حروف من الملف فانا سوف نقوم بالتالي :

```
Dim F As Integer
Dim S As String

S = Space(10)
F = FreeFile

Open FileName For Binary As F
Get #F, 1, S
Close F
```

لاحظ الذي عملناه في المتغير S قمنا باستخدام الدالة Space والتي تجعل النص بطول معين من الفراغات ونحن جعلناه هنا بطول 10 حروف فارغة وهو عبارة عن المخزن للبيانات التي سوف نقرأها فاذا اردنا ان نقرأ خمسة حروف فقط نستخدم Space وبين قوسين الرقم ٥ ثم بعد ذلك فتحنا الملف واستخدما العبارة Get للقراءة من الموقع رقم ١ بطول المتغير S والذي هو هنا ١٠ حروف اي ان النص الذي سيكون في المتغير S بعد العملية السابقة هو "على الله " وذلك الان القراءة من بداية الملف بطول عشرة حروف واذا اردنا ان نقرأ الملف كله نستخدم التالي:

```
Dim F As Integer
Dim S As String

F = FreeFile

Open FileName For Binary As F
S = Space( LOF(F) )
Get #F, 1, S
Close F
```

لاحظ اني استخدمت طول المتغير الذي يخزن البيانات هو LOF(F) وهو كما عرفنا فهو طول الملف الذي رقمه F ويمكنك ان تلاحظ اني نقلت العبارة Space من الخارج الى بعد العبارة Open وقبل العبارة Close وذلك الان الامر LOF يعني ماهو طول الملف F ولو جعنا العبار قبل السطر Open فان الملف F سيكون لم يفتح بعد لذى لن يعرف ماهو F وسعطي خطأ لدى وضعت العبارة بعد السطر الذي يفتح الملف F لذى فان العملية سوف ينتج عنها ان المتغير S سوف يكون بطول الملف كله وبعد ذلك استخدمنا Get للقراءة من الموقع الاول بطول S ينتج ان محتويات الملف كاملاً سوف يتم وضع نسخة منها في المتغير S بعدها يمكن ان نعرض محتويات المتغير S في مربع نص مثلاً ... كالتالي :

```
Text1.Text = S
```

اخوتي يجب ان تجربوا عمليات مختلفة على الملفات حتى تتكون لديكم فكرة قوية عن البرمجة الصحيحة فلا يجوز ان يجهل احدا كيف يتعامل مع الملفات بواسطة الكود وبعيداً عن عملية نقل الملفات بواسطة الويندوز ان التعامل مع الملفات مهم جداً لعمليات التخزين او التشفير او التجزئة او حفظ واسترجاع البيانات فقد تحتاج يوماً ان تبني لبرنامجك نوع معين من الملفات بامتداد خاص بك وتقوم بتخزين بيانات خاصة بك فيه واجرا العمليات المختلفة عليها واطن ان الامر يستحق بعض العناء في تعلمه و لا يقل اهمية عن موضوع درسنا في الربط بين الاجهزة لذى قم بالبحث في المنتدى عن اي مواضيع او اكواد قد تفيدك في التعامل مع الملفات بينما اقوم انا بالتحضير للدرس الذي يليه ..

الدرس السابع : التعامل مع الملفات (ارسال واستقبال)

علمنا سابقاً كيف يمكن ان نفتح ملف ونقرأ ما فيه بشكل كامل LOF ويمكن ان يتبادر الى الذهن ان الموضوع منتهي وبما اننا نعرف كيف يمكن ان نرسل النصوص عن طريق SendData فنحن نستطيع بذلك ان نقوم بفتح الملف ثم نخزن جميع محتوياته في المتغير S ثم بعد ذلك نستخدم الامر SendData لارسال البيانات وعندما تصل الى الطرف التالي يقوم بفتح ملف جديد ثم يخزن البيانات التي وصلته داخله عن طريق Put بدلاً من عرض الرسائل كما كان في برامج المحادثة السابقة ... اليس كذلك ؟

طبعاً الاجابة هنا نعم ... ولكن الامور لا تجري بهذا الشكل في الظروف الحقيقة حيث ان سرعة الانترنت او الشبكات ليست كما تراه الان عندما تجرب البرامج داخل كمبيوترك لذى فقد تقوم بفتح ملف كبير ثم ترسل بياناته دفعة واحدة وهذا خطأ كبير يرتكبه البعض في ارسال الملفات وان كان يعمل بعض الاحيان فهو لن يعمل في احيان اخرى ... خذ مثلاً نحن جميعاً نستخدم الماسنجر .. فكم مرة ظهرت لك الرسالة تعذر ارسال الملف ؟؟؟ ذلك على الرغم من ان الماسنجر هو برنامج صممه مايكروسوفت ! ... فلماذا هذا العيب الذي يظهر فيه احياناً ؟ طبعاً هناك امور مؤثرة كثيرة واهمها اختلاف سرعة التوصيل ففي بعض الاحيان عندما نقوم بتحميل ملف من الانترنت نجد ان سرعة التحميل هي مثلاً 5 KB في الثانية وقد تستغرب اذا قلت لك اني في احدى المرات وجدت ان سرعة التحميل لا تتعدى 7 بايت في الثانية !!! ... ولك ان تتخيل كم من الوقت يحتاج نقل ملف بطول 1 ميجا بايت بهذه السرعة !! مثلاً اذا كانت سرعة النقل هي 1 كليوبات في الثانية فهذا يعني انك تحتاج الى اكثر من 17 ساعة لنقل ملف بطول واحد ميجا بايت

مما سبق تظهر الحاجة الى استخدام تقنية اخرى لتلافي هذه المشكلة وهي في هذه الحالة تقسيم الملف الى دفعات ثم بعد ذلك نرسله بشكل متتالي الى الطرف الاخر مع تمكين الطرف الاخر من ان يلغي الاستقبال في اي لحظة يريد ... وهذا يضمن اولاً عدم الضغط على الشبكة ثم ان عملية الارسال تكون سريعة ذلك ان معالجة كمية كبيرة من البيانات تأخذ وقت وجهد كبيرين على الجهاز مما يشكل ضغط يوتر على عمل باقي البرامج وقد لا يتم استلام البيانات اصلاً . بينما الارسال المتتالي يسمح للنظام بان ينظم العمليات بشكل سليم مع عدم اعاقه عمله ...

الان اصبح لدينا اسبابنا كي نرسل الملف بشكل متتالي بدلاً من ارساله دفعة واحدة وبقي ان نعرف كيف نترجم ذلك برمجيّاً ...

دعنا الان نفترض حجم ملف تخيلي مثلاً 18 بايت واردا ارساله بطول ستة بايت في كل مرة يعني ذلك اننا يمكن ان نستخدم دوران عن طريق For Next او Do While او ما شابه ونجعله يدور ثلاث مرات في كل مرة نقرأ من الملف بيانات بطول ستة بايت عن طريق S=Space(6) ثم الامر Get وبعدها نرسل عن طريق SendData S هذا جميل ... ولكن ماذا اذا كان طول الملف هو 20 وليس 18 فاننا هنا سوف نعمل شيء قريب من هذا :

```

Dim S As String
Dim I As Integer
Dim F As Integer

F = FreeFile
Open FileName For Binary As F

For I = 1 To LOF(F) \ 6
    S = Space(6)
    Get #F, , S
    wsk.SendData S
    DoEvents
    Print S; "END"
Next

If LOF(F) Mod 6 Then
    S = Space(LOF(F) Mod 6)
    Get #F, , S
    wsk.SendData S
    DoEvents
End If

Close F

```

إذا دققنا في الكود السابق سوف نجد اننا استخدمنا اشياء غريبة بعض الشيء مثل $\text{LOF}(F) \setminus 6$ وكذلك $\text{LOF}(F) \text{ Mod } 6$ وكل ذلك بسبب اننا نريد ان نعرف كم حجم البيانات المناسب الذي نريد قراتها من الملف خذ مثلاً في مثالنا السابق كانت البيانات بطول ١٨ بايت ونريد ان نقرأ بطول ستة لان العدد ١٨ يقبل القسمة بدون باقي على ٦ فان الاجزاء التي سوف نقرأها ستكون كالتالي $6 + 6 + 6 = 18$ وهذا يعني امكانية استخدام $S = \text{Space}(6)$ لمدة ثلاث مرات وينتهي العمل لكن اذا كان الطول هو ٢٠ فان ذلك يعني $6 + 6 + 6 = 20$ اي اننا يجب ان نقرأ بطول ستة ثلاث مرات اما المرة الاخيرة فاننا يجب ان نقرأ بطول ٢ وليس ستة يعني $S = \text{Space}(2)$ وذلك لاننا اذا قرأنا المرة الاخيرة بطول ستة فان حجم الملف الجديد سوف يكون ٢٤ وليس ٢٠ وهذا يعني ان الملف الجديد لن يكون نسخة طبق الاصل من الملف المرسل وهذا ما لا نريده ولكي نتجنب ذلك يجب ان نعرف كم مرات نقرأ بالطول العادي (سته) ثم ما هو الطول الاخير المتبقي كي نجعل طول المتغير S يتناسب معه ... اذا امعنا النظر في $6 + 6 + 6 = 20$ نجد اننا نقرأ ثلاث مرات بشكل طبيعي وهذا يمكن ان نعرفه عن طريق قسمة طول الملف $20 / 6 = 3$ على طول الاجزاء التي نريد ارسالها وهي ٦ ولكن القسمة تكون بدون باقي وهذا سوف يعطي ثلاث لان القسمة مع الباقي تعطي ٢,٣٣٣٣٣٣٣ اي ثلاثة واجزاء كسرية ولكي نقسم في الفيجوال بيسك بدون باقي نستخدم (\) وليس الاشارة العادية للقسمة (/) خذ هذا المثال :

$12 / 8 = 1.5$	مع الباقي
$46 / 10 = 4.6$	مع الباقي
$12 \setminus 8 = 1$	بدون باقي
$46 \setminus 10 = 4$	بدون باقي

وبذلك نكون قد عرفنا عدد المرات التي نقرأ فيها بطول عادي وهو سته بايت وانظر الى الكود ستجد اني استخدمت للدوران For من واحد وحتى $\text{LOF}(F) \setminus 6$ وذلك لان طول الملف يأتي مخزون في LOF كما عرفنا سابقاً وهذا يضمن ان يتغير العدد بتغير طول الملف فاذا كان الملف طوله $\text{LOF} = 15$ فان الناتج من قسمته بدون باقي على سته هو ٢ واذا كان طول الملف ٢٤ فان الناتج سيكون ٥ وبهذا نضمن مرونة الكود بشكل يناسب كل الملفات ...

وبقي ان نحسب الجزء الاخير من الملف ويمكن حسابه عن طريق الامر Mod حيث انه يعود بباقي القسمة اي ان (٢٠ ثم mod ثم ٦) يعطي الناتج ٢ ويمكن بعدها ان نجعل طول المتغير $S =$

Space(LOF(F) Mod 6) وهو ما عملناه بعد انتهاء الدوران For Next في الكود السابق حيث قمنا بالتالي :

```
If LOF(F) Mod 6 Then
  S = Space(LOF(F) Mod 6)
  Get #F, , S
  wsk.SendData S
  DoEvents
End If
```

حيث قمنا بالتأكد انه يوجد باقي في السطر If وذلك لانه اذا لم يبقى باقي فلاداعي لتأكد مثل اذا كان طول الملف هو ١٨ فان ناتج القسمة هو ٣ والباقي من القسمة هو ٠

ثم بعد ذلك قمنا بحجز طول المتغير S بحسب الجزء المتبقي من الملف ثم قرانا البيانات من الملف وارسلناها باستخدام...SendData

في مثالنا السابق نلاحظ اننا استخدمنا الرقم ٦ بايت ونحن في الحقيقة لن نستخدم هذا الرقم ولكننا سوف نستخدم طول مختلف ممكن نستخدم طول 2 كليوبايت او اربعة كليوبايت وهي ارقام مناسبة ويمكنك تغييرها اذا رايت ان ذلك يتناسب مع سرعة خدمة الشبكة الاعتيادية لديك ونحن سوف نستخدم الطول ٤ كليوبايت في دروسنا القادمة اي ٤٠٩٦ بايت

واخيراً نجد اننا لم نجعل كود يمكن ان يوقف العملية في المنتصف اي قبل ان تنتهي حيث ان الدوران يستمر حتى نهاية الملف ولكن هذا غير مقبول اذا كان الملف كبيراً واراد المستخدم الغاء العملية لذي سوف نضيف شرط يتأكد من العملية لا يراد لها ان تنتهي بحيث يتمكن المرسل والمستقبل من انتهاء عملية التراسل متى ارادوا ذلك

والمثال على هذا الدرس جاهز الان ولكن بقي ان اكتب له شرحاً مناسباً حيث انه معقد بعض الشيء ويعتمد على كثير من الرسائل التي يتبادلها العميل والخادم كي تتم العملية بشكل سليم

تكملة ...

بسم الله الرحمن الرحيم

المثال الذي نضعه هنا هو مثال اولي لعملية ارسال واستقبال الملفات بحيث نستمر بعد ذلك في الدروس القادمة بتطويره كي نصل الى عملية ارسال واستقبال اكثر موثوقية ...

سوف نبدء الان في مناقشة العمل الذي يقوم به برنامج العميل وهو في درسنا هذا المسؤول عن ارسال الملفات بينما يقوم الخادم بدور المستقبل ...

يوجد في برنامج العميل اربعة ازرار هم : تحديد الملف ، اتصال ، طلب ارسال و إلغاء الارسال ...

في البداية يقوم العميل باختيار الملف الذي يريد ارساله من القرص الصلب عن طريق الزر تحديد الملف فيظهر له مربع الحوار فتح فيختار الملف المراد ارساله ... ويجب طبعاً النقر على الزر اتصال حتى ينشأ الاتصال بين الخادم والعميل ويتأكد من ان الحالة هي "متصل" ...

بعد ذلك ينقر على الزر طلب ارسال وبهذا يتم ارسال رسالة الى الخادم محتوية على الامر NewFile ثم اسم الملف الذي يراد ارساله ... طبعاً اسم الملف سيكون خالياً من المسار بحيث اذا كان الملف في برنامج العميل هو

```
C:\Data\All Programs\Game.exe
```

فاننا لن نقوم بارسال العنوان كاملاً وذلك لاننا لا نضمن ان يكون هذا المسار موجود في برنامج الخادم حيث ان البرنامج سوف يعملان في الحقيقة في جهازين منفصلين وبالتالي الاسم المرفق في مثالنا السابق سيكون Game.exe فقط وسوف تكون الدالة FileNameOnly هي المسؤلة عن فصل الاسم عن المسار وتجدها في برنامج العميل ... وبذلك تكون الرسالة المرسله الى الخادم عند النقر على الزر طلب الارسال "NewFileGame.exe" حيث ان السبعة الحروف الاوالية هي للامر والباقي لأسم الملف بحيث يقوم الخادم بالفصل عند الاستلام كالتالي:

```
Cmd="NewFile"  
Msg="Game.exe"
```

وذلك عن طريق الاوامر Left و Mid كما تعودنا سابقاً والمهم هنا ان العميل بمجرد استقبال هذا الامر NewFile يقوم مباشرة بعرض رسالة الى المستخدم يخبره ان العميل يريد ارسال ملف اليه فهل يقبل استلام الملف ام لا (كما هو الحال في الماسنجر عند تناقل الملفات بين المستخدمين) .. ويكون اما المستخدم اما القبول او الرفض فاذا نقر على الزر موافق يتم ارسال رسالة الى العميل هي OkSend تخبره بان يبدأ الارسال او في حالة الرفض يتم ارسال الرسالة NoSend ومن هاتين الرسالتين يكون على العميل معرفة الخطوة التالية حيث انه اذا استقبل الرسالة OkSend يعرض رسالة الى المستخدم بانه تم قبول ارسال الملف ثم يبدأ بعدها مباشرة بارسال الملف اما اذا استلم الرسالة NoSend فانه يقوم بعرض رسالة تفيد المستخدم ان طلب الارسال تم رفضه من قبل الخادم

والان سوف نناقش سوياً ماذا يحدث اذا تم قبول ارسال الملف ... حيث انه يقوم العميل بعد ان يقبل الخادم استقبال الملف باستدعاء الاجراء SendFile والذي سوف يكون المسؤول عن كود الارسال وهو كالتالي :

```
Private Sub SendFile(FileName As String)  
    Dim S As String  
    Dim F As Integer  
    Dim P As Long  
  
    Const Size = 2048  
  
    If wsk.State <> 7 Then  
        MsgBox "لا يوجد اتصال"  
        Exit Sub  
    End If
```



```

Command1.Enabled = False
F = FreeFile
Open FileName For Binary As F

For P = 1 To LOF(F) \ Size
    S = Space(Size)
    Get #F, , S

    If Sending = False Then
        GoTo Bye
    End If

    If wsk.State = 7 Then
        wsk.SendData "NewPart" & S
    Else
        MsgBox "لقد تم قطع الاتصال"
        GoTo Bye
    End If

    DoEvents
Next

If (LOF(F) Mod Size) > 0 Then
    S = Space(LOF(F) Mod Size)
    If wsk.State = 7 Then
        Get #F, , S
        wsk.SendData "NewPart" & S
        DoEvents
    Else
        MsgBox "لقد تم قطع الاتصال"
        GoTo Bye
    End If
End If

If wsk.State = 7 Then
    wsk.SendData "EndFile"
    DoEvents
End If

Bye:
Close F
Command1.Enabled = True
End Sub

```

وسوف نناقشه معاً بتروي الا وعلى بركة الله

لاحظ اولاً اننا حددنا حجم القطع التي سوف نرسالها ووضعناها في الثابت Size حيث اننا وكما سبق ان قلنا ان الارسال لن يتم دفعة واحدة وانما على دفعات ونحن هنا حددنا حجم الدفعات بـ ٢٠٤٨ بايت والذي يعني ٢ كيلوبايت ثم قمنا بعد ذلك بالتأكد من انه يوجد اتصال ام لا فان لم يكن هناك اتصال فلا داعي لارسال الملف ونكتفي بعرض رسالة على المستخدم بعدم توفر اتصال مع الخادم في هذه اللحظة....

ثم قمنا بعد ذلك بفتح الملف الموجود اسمه في المتغير FileName وبعدها بدأنا في الدوران For Next....والذي سوف يقوم بالارسال على شكل دفعات حيث نقوم اولاً بقراءة الدفعة الاولى من الملف عن طريق Get ونخزن البيانات التي نقرأها في المتغير S ثم بعد ذلك نفحص المتغير Sending فاذا كان False نوقف عملية الارسال ثم نذهب الى السطر Bye الموجود في الاسفل وفيه يتم اغلاق الملف والخروج بعدها.. والمتغير Sending هو الذي يحدد ان الخادم يريد الاستمرار في استلام الملف حيث انه في الملفات الكبيرة تستغرق العملية وقتاً طويلاً وإذا اراد الخادم ان يلغي العملية يقوم بالضغط على الزر

"إلغاء الاستلام" فيقوم بإرسال رسالة "Cancel" الى العميل والذي بدوره عند استلامها يجعل قيمة المتغير Sending=False وبما ان هذا المتغير يتم فحصه دائماً قبل إرسال أي دفعة فانه اذا أصبحت قيمته False تتوقف العملية مباشرة ويتم عرض رسالة للعميل بان الخادم اوقف العملية ...

نعود الان لاستكمال الاجراء SendFile فكما قلنا بعد التأكد من ان المتغير Sending لا يساوي False نقوم بإرسال الدفعة بواسطة SendData S وكما تعودنا دائماً اننا لا نقوم باستخدام الامر SendData قبل ان نتأكد من حالة الاتصال هي V أي متصل نقوم مباشرة بالارسال ولاحظ اننا نرسال البيانات مسبقة بالامر NewPart والذي يمكن الخادم من معرفة ان هذه الرسالة تحتوي على جزء آخر من اجزاء الملف فيقوم عند استلامها بفتح الملف ثم اضافة البيانات اليه كما عرفنا في درس التعامل مع الملفات ثم انتظار القسم التالي وتستمر هذه العملية (ارسال واستقبال الدفعات) ما لم ينقر احد المستخدمين على الزر الغاء والذي يحول قيمة المتغير Sending=False وبما ان هذا المتغير يتم فحصه دائماً لذي فبمجرد عدم ايجابيته (False) فان العملية بالكامل يتم الغائها

وبعد ان يتم ارسال آخر جزء من الملف بنجاح يقوم العميل بعدها بإرسال الرسالة EndFile والتي تعني ان الملف تم ارساله بالكامل وعندما يستلم الخادم هذه الرسالة يقوم بعرض مربع حوار يفيد المستخدم بان العملية انتهت بنجاح.....

.....

في هذا المثال تطرقنا الى ارسال الملفات بصورة نموذجية على اساس ان الاتصال يكون في احسن حالاته ولكن ذلك غير وارد في كل الاحيان لانه كما ناقشنا سابقاً توجد اختناقات في الشبكة الحقيقية لذي فان المثال سوف يعمل في جهازك او في شبكة داخلية بسيطة ومع ذلك قد تنشأ مشاكل في الارسال في بعض الحالات لذي سوف نناقش مستقبلاً بعض الاساليب المتبعة للوصول الى نتائج افضل في اغلب الاحيان كما ان هذا المثال مجرد من متطلبات التناقل الاساسية كاشريط حالة يوضح سير العملية وماهو المقدار التي تم ارساله حتى الان وغيره وهذا ما سوف نتطرق اليه لاحقاً بإذن الله تعالى والمهم هنا التأكد من استيعاب العملية والكود الموجود في هذا الدرس بشكل كامل بحيث يصبح لدينا اساس قوي يمكن ان نتمتع بعدها في الموضوع معتمدين على المعلومات المناقشة هنا.....

ملاحظة :

ان الملف الذي سوف يتم ارساله يتم استقباله في برنامج الخادم بنفس المجلد الذي يوجد فيه برنامج الخادم أي انك اذا اردت التأكد من نجاح عملية الارسال ابحث عن الملف الذي تم نقله في مجلد برنامج الخادم وسوف نضيف مستقبلاً امكانية تحديد المكان الذي نريد ان نستقبل الملف فيه الى برنامج الخادم بإذن الله تعالى

إضافات على برامج ارسال الملفات

اما الان واستمراراً في مشوارنا الذي بدئناه معاً دعونا نقوم بتطوير طريقتنا لارسال الملفات وسوف نناقش الان وبإذن الله تعالى كيفية اضافة شريط حالة (ProgressBar) لعملية الارسال كي يعطي كلاء الطرفين فكرة عن المرحلة التي وصلت اليها العملية ولقد استخدمت لهذا الغرض شريط للحالة قمت بتصميمه من قبل [تجده على هذا الرابط](#) ويمكنك بالطبع استخدام شريط الحالة التي تقدمه مايكروسوفت مع الفيجوال بيسك وانا لم استخدم شريط الحالة الذي صممته هنا الا لانه يحتوي على النسبة المئوية التي تبين سير عملية ارسال الملف وهو ما لا تجده في الشريط الافتراضي الذي يأتي مع الفجوال بيسك ...

والمهم في الامر ان شريط الحالة يحتوي على ثلاث خصائص مهمة هي Min و Max و Value:

Max	هو عبارة عن آخر درجة في الشريط
Min	وهو عبارة عن اول درجة في الشريط
Value	عبارة عن الدرجة الحالية

والفكرة بشكل مبسط هي ان نجعل القيمة $Min=0$ ونجعل القيمة Max تساوي حجم الملف الكلي اي $LOF(F)$ ونجعل القيمة $Value=0$ في البداية وبعدها عند ارسال كل جزء من اجزاء الملف نقوم بزيادة القيمة Value بحسب حجم الجزء الذي تم ارساله من الملف وبما ان الجزء الذي يتم ارساله يتم تخزينه في المتغير Msg فان الزيادة للقيمة Value تكون $Len(Msg)$ اي حجم المتغير Msg

وبقي الان نعرف كيف نطبق ذلك في برنامجنا

نحن نعرف انه في الدرس السابق عندما يقبل الخادم استلام الملف فان العميل يقوم باستدعاء الاجراء المسمى SendFile وهو المسؤول عن عملية ارسال الملف على شكل دفعات وعلينا الان تعديل هذا الاجراء كي يتناسب مع الاضافات التي نريدها وهي شريط الحالة الجديد ... ولذلك ينبغي لنا قبل ان نرسل الجزء الاول الى الخادم ان نرسل رسالة توضح له ماهو طول الملف الذي نريد ارساله حتى يغير الخادم الخاصية Max لشريط الحالة عنده بما يتناسب مع طول الملف الذي سوف يتم ارساله حيث نقوم بارسال رسالة باسم "TheSize" متبوعة بحجم الملف اي $LOF(F)$ وعندما يستلمها الخادم يعرف طول الملف الذي سيتم ارساله فيقوم بجعل الخاصية Max لشريط الحالة مساوية لطول الملف

وبعد ذلك في كل مرة يتم ارسال جزء جديد من الملف يقوم كلاء البرنامجين الخادم والعميل بزيادة قيمة الخاصية value التابعة لشريط الحالة مما يعطي صورة واضحة عن تقدم عملية الارسال.....

طبعاً ان كل ما سبق يعتمد على مدى فهمك للتعامل مع شريط الحالة او ال ... ProgressBar اما مسألة الارسال فهي بسيطة وكما اعتدنا عليها سابقاً لذى في حالة انك وجدت صعوبة في فهم التعامل مع شريط الحالة راجع الرابط السابق الذي وضعته لك كي ترى كيف يتم استخدامه بشكل مبسط ثم بعد ذلك قم بتفحص الدرس الجديد كي ترى كيف استخدمناه في درسنا هذا

وسيكون الدرس التالي بإذن الله تعالى يتعلق بإضافة اشيء بسيطة لبرنامجنا من شأنها ان تحسن طريقة ارساله واستلامه وخاصة للملفات الكبيرة

طريقة اخرى لارسال الملفات

السلام عليكم ورحمة الله وبركاته....
إخوتي بعد أن نقاشنا فيما سبق طريقة إرسال الملفات بشكل مبسط لا يخلو من العيوب سنحاول معاً أن نحسن من طريقتنا السابقة بحيث نتمكن من زيادة نسبة نجاح عملية الإرسال.... طبعاً إن ما سوف نضيفه هنا لا يضمن في كل الحالات أن يتم النقل بصورة كاملة وذلك نظراً للعوامل المختلفة المؤثرة عليها لذا هي إحدى محاولاتنا لزيادة الكفاءة...

والآن نحن نعرف (تربيع). فرق بين النظرية وبين الواقع حيث إنني إذا سألت أحدكم ما هو مقدار الجاذبية الأرضية فإنه سيجيب مسرعاً (٩,٨ متر لكل ثانية تربيع) .. ولكن هل هذا صحيح فعلاً .. في الواقع هو صحيح من الناحية النظرية ولكن واقعياً إن معدل جاذبية الأرض يتأثر بعدة عوامل أخرى مما يجعله يختلف من منطقة لأخرى فقد يكون أعلى أو أدنى من تلك القيمة...

وهذا بالضبط ما يحدث في عملية النقل فهي تتأثر بسرعة الأجهزة المستخدمة وبحجم الذاكرة في كل منها كما إن لخطوط الشبكة دور كبير ولمقدار الضغط المسلط على الشبكة في الوقت الذي يتم الإرسال فيه... لذا مهمتنا هنا هي محاولة التغلب على هذه الظروف أو على الأقل أخذها بالحسبان في عملية الإرسال لزيادة كفاءتها ونجاحها في أغلب الحالات ...

ولكي نتغلب على مشكلة الذاكرة ويمكن خفض حجم الأجزاء التي سوف نرسلها في كل مرة مثلاً استخدمنا نحن الحجم 4 كيلوبايت حيث جعلنا الثابت Size=4096 ويمكننا أن نجعله واحد أو اثنين كيلوبايت مثلاً Size=2048 أي اثنين كيلوبايت ولك أن تغيرها كما يحلو لك ..

والآن إن الفكرة التي اعتمدناها سابقاً في الإجراء المسؤول على عملية الإرسال SendFile هي أن يتم الإرسال بشكل متتالي وبصورة ثابتة حيث أنه يقوم باستخدام دوارن For...Next للإرسال دون مراعاة نجاح عملية الإرسال في كل مرة ولذا إذا حدثت مشكلة في إرسال أي جزء فإنه لن يأخذ ذلك في الحسبان بل سوف يقوم بإرسال الجزء الذي يليه وهو شيء غير صحيح في الواقع العملي

لذا سوف نستخدم طريقة أخرى وهي أن نصمم إجراء فرعي يجعل البرنامج ينتظر لمدة معينة من الثواني قبل أن يقوم بإرسال الجزء التالي ولنقل ٢٠ ثانية وبذئ نعطي فرصة لعملية الإرسال كي تأخذ وقت كافٍ والإجراء يكون بشكل يشبه التالي :

```
Private Sub Wait20Seconds()  
    Dim T As Single  
  
    T = Timer  
    Do Until (Timer - T >= 20)  
        DoEvents  
    Loop  
  
End Sub
```

والإجراء كما هو واضح يقوم أولاً بمعرفة الوقت الحالي ووضعه في متغير T ثم يقوم بالدوارن إلى أن يكون الفرق بين وقت بداية الإرسال T والوقت الحالي هو ٢٠ ثانية ثم يخرج وهو يحقق فترة انتظار لمدة عشرين ثانية أي أنك إذا وضعت الكود التالي في زر أمر مثلاً :

```
Private Sub Command1_Click()  
    MsgBox "الانتظار"  
    Wait20Seconds  
    MsgBox "بعد الانتظار"  
End Sub
```

فإن الذي سوف يحصل أنه سيتم عرض رسالة فيها العبارة "قبل الانتظار" وذلك بمجرد الضغط على الزر ثم بعد ما يقارب عشرين ثانية سيتم عرض الرسالة أخرى فيها العبارة "بعد الانتظار" ونلاحظ أنه على الرغم من السطور الثلاثة هي سطور متتالية ويجب تنفيذها بشكل متتالي وهو ما حصل فعلاً غير أنه عندما بدء الكمبيوتر بتنفيذ أوامر الإجراء Wait20Seconds فإنه اضطر للانتظار لمدة ٢٠ ثانية ثم تنفيذ السطر الذي يليه وهو الرسالة الثانية ...

ونحن نعرف انه في برنامجنا للارسال نستخدم الامر DoEvents بعد الامر SendData كي يوفر وقت بسيط للانتظار ولكننا الان بحاجة لوقت اطول لذي سوف نستبدل الامر DoEvents الذي استخدمناه للانتظار بعد ارسال كل جزء من الملف بالامر Wait20Seconds وبذا نتيح فرصة اكبر للبرنامج كي ينفذ عملية الارسال

قد يتبادر الى الذهن ان ما سبق هو حل عملي وكافي للتغلب على مشاكل الارسال ولكننا في الواقع قد تغلبنا على جزء من المشكلة ولكننا في الحقيقة اضفنا مشكلة اخرى ويمكنني توضيحها كالتالي لنفرض ان الملف الذي نريد ارساله حجمه هو واحد ميغا بايت يعني ١٠٢٤ كيلوبايت فان برنامجنا سوف يرسل الملف على شكل اجزاء بحجم ٢ كيلوبايت اي ان عدد الاجزاء التي سوف يتم ارسالها هي ٥١٢ جزء ولأننا جعلنا الفترة بين كل عملية ارسال لجزء ما والجزء الذي يليه هي ٢٠ ثانية فاننا سوف نحتاج الى مايقارب ثلاث ساعات لكي نرسل هذا الملف!!!!

بينما ارسال ملف بهذا الحجم لا يتطلب كل هذا الوقت الخيالي ... ذلك انه ليست عملية ارسال لجزء من اجزاء الملف تتطلب وقت يقدر بعشرين ثانية بل انه ربما تحتاج بعض الاجزاء الى اقل من ثانية واحدة واجزاء اخرى تحتاج ثانية او اثنتين و و

حيث ان كل جزء يحتاج الى وقت معين بحسب حالة النظام وحالة الشبكة و باقي العوامل المؤثرة التي ذكرناها سابقاً والتي تحدث في نفس وقت ارسال جزء ما من اجزاء الملف ومن هذا ان كل جزء من اجزاء الملف يحتاج الى وقت معين قد يكون اقل بكثير من ٢٠ ثانية او اكثر منها بحسب الظروف المختلفة ...

اذا فأن الذي نحتاج اليه فعلاً هو ان نعرف مقدار الوقت الذي يحتاجه كل جزء ثم نقوم بالانتظار بما يتناسب معه وبذلك لا نهدر الوقت في انتظار عملية ارسال قد تمت فعلاً

ولحل هذه المسألة يمكننا الاستفادة من الحدث SendComplete التابع للاداء Winsock وهذا الحدث ينفذ عندما تنتهي عملية ارسال بيانات معينة اي انك عندما تستخدم الامر SendData لارسال بيانات فانه بعد ان يتم تنفيذ الارسال يتم استدعاء الحدث SendComplete وهو يفيد بان السطر SendData قد تم تنفيذه بنجاح...

لذي سنقوم بجعل الاجراء Wait20Seconds ينتظر لمدة ٢٠ ثانية بشكل قياسي ولكن اذا حصل ان الحدث SendComplete ثم استدعاه قبل انتهاء العشرين ثانية فان هذا يعني انه لاداعي لمزيد من الانتظار ذلك ان الارسال قد تم بنجاح ولذي سنقوم بتعديل هذا الاجراء بحيث ينهي الانتظار في حالة انتهاء الارسال بنجاح حتى وان لم تنتهي الفترة المحددة له وهي ٢٠ ثانية ...

ونحتاج بهذا لتعريف متغير عام نسميه مثلاً OK من النوع Boolean والذي يعني True او False ثم نجعل الكود في الحدث SendComplete كالتالي :

```
Private Sub wsk_SendComplete()  
    Ok = True  
End Sub
```

وهذا يعني ان نجعل قيمة المتغير Ok تساوي True في حالة نجاح الارسال ونقوم بعد ذلك بتعديل الاجراء Wait20Seconds كالتالي :

```
Private Sub Wait20Seconds()  
    Dim T As Single  
  
    Ok = False  
    T = Timer  
    Do Until (Timer - T >= 20) Or Ok  
        DoEvents  
    Loop  
  
End Sub
```

وبذلك ترى اننا غيرنا قيمة المتغير Ok الى False قبل الانتظار ثم قمنا بالانتظار الى ان يتم احد الشرطين وهما ان تنتهي المهلة المحددة وهي ٢٠ ثانية او ان يعمل الحدث SendComplete فيجعل قيمة المتغير Ok تساوي True وبذلك ينتهي انتظارنا عندها لان الارسال يكون قد تم فعلاً ولاداعي حينها للتأخير

طبعاً هذه الطريقة هي محاولة لتغلب على مشكلتنا في الارسال بالنسبة للملفات الكبيرة وغيرها وهي وان كانت طريقة اكثر قوة من سابقتها فهي لا تعني سلامة الارسال في كل الاحوال ... وقد جربتها انا مع بعض الملفات الكبيرة وعملت بشكل جيد والحمد لله تعالى وسوف نتطرق بعدها الى اضافة بعض المميزات اليها لاحقاً ان شاء الله تعالى

ملاحظة بسيطة

نحن كنا في السابق نستخدم الاجراء Wait20Seconds والذي يقوم بالانتظار لمدة عشرون ثانية او حتى يكون المتغير OK يساوي True وهو ما يحدث نتيجة الحدث SendComplete بالشكل التالي :

```
Private Sub wsk_SendComplete()  
    Ok = True  
End Sub
```

وما نحتاجه نحن في هنا ان نجعل المتغير OK تكون قيمته True عند انتهاء الخادم من حفظ البيانات (او الدفعة الاخيرة) الى الملف وبالتالي فهو يكون على استعداد لاستقبال دفعة اخرى ولذى نقوم بجعل الخادم عند انتهائه من حفظ كل الجزء من الملف بارسال رسالة الى العميل كي يغير قيمة المتغير OK الى True وبالتالي يتم ارسال البيانات الجديدة وجعلت الرسالة باسم "OKTrue" وبمجرد ان يستلمها العميل يجعل OK=True وهو التغير الوحيد في العميل مما يسبب استكمال ارسال الملف بالتدريج ..

ومن عيوب هذه الطريقة هي البطء النسبي ذلك ان العميل بعد ارسال كل جزء من البيانات يظطر الى الانتظار حتى يستلم من الخادم اشعار باستعداده لاستقبال الجزء التالي وبدى يصبح العمل على شكل تبادل تراسل مستمر بين العميل والخادم ولكنها طريقة تعطي وثوقية اكبر من ناحية سلامة عملية ارسال واستقبال الملفات ...

ونظراً لهذا التبادل المستمر في الرسائل افضل في المشاريع الكبيرة ان يتم الاعتماد على اكثر من اداة وينسوك بحيث تكون واحدة خاصة فقط بعملية الاسال والاخرى لتنفيذ عمليات تبادل الاوامر الاخرى التي يؤديها البرنامج ولا تعتمد على ارسال الملفات ...

وكمثال تجد انك حين تقوم بارسال او استقبال ملف على الماسنجر تبقى لديك القدرة على الاستمرار في الحديث اثناء تقدم العملية طبعاً يمكنك استخدام نفس الاداه (وينسوك) للتشات والارسال في نفس الوقت ولكن ذلك قد يسبب تداخل اوامر الارسال واوامر التشات وهو رأي شخصي مبني على الاعتقاد فقط

المهم هنا اننا بعد ماسبق لانريد ان تتكون صورة باننا قد بنينا برنامج ارسال متكامل بل نقول نحن فقط عملنا جهدنا لانجاح العملية بشكل يمكننا من ارسال ملفاتنا في معظم الاحيان وليس جميعها بشكل سليم ان شاء الله تعالى وقد اعتمدت فيما سبق على تجارب شخصية لذى فاني اقول اذا كنا قد عرفنا الكثير عن العمليات السابقة فقد خفي عنا الكثير ايضاً واتمنى للجميع حظاً موفقاً مع هذه الاداه

ولكنني اظن انه بقيت اجزاء مهمة يجب ان نخوض فيها وهي عملية حماية انفسنا من الهكر والذي اعتقد انهم مهما عملوا فهم مبرمجون مثلنا واذا كان لهم عقول سوداء فلنا عقول مسلمة تأبى ان تضر الغير مع عدم عجزها ابداً عن الدفاع عن نفسها ياذن ربها

وسوف احاول توضيح بعض الطرق التي قد لاتوفر حماية كاملة ولكنها على الاقل لن تدعنا لقمة سائغة ان شاء الله تعالى ...

واتمنى ان يعيني الله واياكم على رفع شأن ديننا وامتنا الاسلامية عالياً ان شاء الله تعالى ...

الدرس الثامن

السلام عليكم ورحمة الله وبركاته ...

دعونا الان نبين امراً

نحن فيما سبق رأينا كيف يمكن ان ننقل ملفاً باستخدام الوينسوك وقد اعاننا الله على ذلك ... وهي طبعاً ليست الطريقة المثلى ولكنها افضل ما توصلنا اليه سوياً ...

وقد رأيت اهتمام بعض الاخوة بنقل صورة عن سطح المكتب وسوف نناقش هنا الامر بشكل نظري لتوضيح الفكرة ... فلنقوم بذلك يجب ان نفكر بسطح المكتب كصورة عادية اذ ان العملية تتم بالطريقة التالية :

- ١ - ارسال رسالة الى الخادم كي يأخذ صورة لسطح المكتب.
- ٢ - يقوم الخادم بأخذ الصورة ويحفظها في ملف ..
- ٣ - يقوم الخادم بأرسال الملف الى العميل ..
- ٤ - يقوم العميل بتحميل الملف وعرضه في مربع صورة..

التوضيح:

لكي نقوم بأخذ صورة لسطح المكتب يمكن ذلك بعدة طرق ايسرها استخدام الدالة keybd_event وهي دالة API تستخدم بغرض اجبار الكمبيوتر على ارسال رسالة الى النظام بأن مفتاح معين قد تم الضغط عليه .. واغلبنا يعرف ان الزر Print Screen الموجود في اعلى لوحة المفاتيح يمكن استخدامه لنسخ صورة سطح المكتب ... اليس كذلك؟؟؟

على العموم ... لمن لايعرف ذلك يقوم بالتالي اولاً قم بالضغط على المفتاح Print Screen الموجود في اعلى يمين لوحة مفاتيح الكمبيوتر ويقوم حينها النظام بأخذ صورة للشاشة متضمنة كل ماتراه انت في تلك اللحظة على الشاشة ويضعها في ذاكرة الكمبيوتر ثم بعد ذلك قم بالانتقال الى برنامج الرسام التابع للويندوز ومن القائمة تحرير اختار لصق ... سوف تجد صورة سطح المكتب بكل ما يحتويه امامك على شكل صورة يمكنك تحريرها...

ونحن هنا نستخدم نفس الفكرة ففي الاول نقوم بجعل البرنامج يخبر الويندوز بانه قد قام احدهم بالضغط على الزر PrintScreen وذلك باستخدام الدالة keybd_event وحينها ينسخ الويندوز صورة سطح المكتب الى الذاكرة (الحافظة او ... Clipboard)

وبعد ذلك نقوم باستخدام الامر SavePicture لكي نحفظ الصورة الموجودة في الحافظة الى القرص الصلب وبهذا يكون لدينا ملف فيه صورة لسطح المكتب وبعدها يمكننا ارساله الى العميل كما نفعل دائماً مع اي ملف عادي ... وعندما ينتهي ارسال الملف يقوم العميل بتحميل الملف الى مربع صورة كي يراه المستخدم وذلك عن طريق الامر LoadPicture الذي يستخدم لتحميل صورة من ملف الى مربع الصور ...

سوف تجدون في الاسفل برنامج بسيط يقوم بعمل وضع صورة سطح المكتب في ملف صورة ومن ثم يقوم بعرض هذا الملف ... وهو كالتالي :

```
Private Declare Sub keybd_event Lib "user32" (ByVal  
bVk As Byte, _  
ByVal bScan As Byte, ByVal dwFlags As Long, ByVal  
dwExtraInfo As Long)  
Private Const VK_SNAPSHOT = &H2C  
  
Private Sub Command1_Click()  
keybd_event VK_SNAPSHOT, 0, 0, 0  
DoEvents
```

```

SavePicture Clipboard.GetData(), App.Path &
"\Desktop.bmp"

Set RgheedViewBox1.Picture = LoadPicture(App.Path
& "\Desktop.bmp")
End Sub

```

في البداية تعريف الدالة keybd_event وكذلك الثابت VK_SNAPSHOT وهو عبارة عن رقم الزر
PrintScreen حيث ان لكل زر رقم معين ...

وفي السطر الاول في الشفرة التي في زر الامر نقوم بأخذ صورة لسطح المكتب ثم نستخدم الامر
DoEvents لكي ننتظر حتى انتهاء العملية ...

بعدها نستخدم الامر SavePicture لحفظ الصورة التي في الحافظة Clipboard.GetData الى داخل
الملف Desktop.bmp

بعد ان انتهينا من حفظ الصورة في الملف نقوم بعرض الصورة في مربع الصور.... RgheedViewBox1

لاحظ ان مربع الصور التي استخدمته لعرض الصورة هو ليس مربع الصور الاعتيادي PictureBox وإنما هو
عبارة عن اداة بسيطة لعرض الصور وهي تشبه مربع الصور العادي مع بعض الاضافات حيث ان الاداة
تمتلك ميزة ظهور اشرطة تمرير اذا كانت الصورة كبيرة وبالطبع فالصورة كبيرة لهذا استخدمت انا هذه
الاداة وبممكنك استخدام ال PictureBox العادي طبعاً

واذا اردت الحصول عل مميزات هذه الاداة الجديدة يمكنك الاطلاع على هذا [الرابط](#) :

<http://www.vb4arab.com/vb/showthread.php?postid=1000032165&t=6859>

طبعاً ان ما سبق كان للتوضيح فقط وفي حالة كونك تريد ارسال الملف باستخدام الوينسوك يفضل دائماً
ان يكون حجمه صغيراً لهذا من الافضل كثيراً ان تحول الصورة التي حصلنا عليها لسطح المكتب من
التنسيق bmp الى jpg حيث ان حجم الملف مثلاً قبل التحويل هو ١,٣٧ ميغابايت اي حوالي (١٤٠٣
كيلوبايت) بينما بعد تحويلها الى تنسيق jpg فانه يصبح تقريباً ٤٨ كيلوبايت وهو حجم صغير جداً
مقارنة مع سابقة وبذلك يصبح الارسال اسرع بكثير ...

وسوف اتطرق لاحقاً الى كيفية تحويل الصورة من bmp الى jpg والعكس ...

ان شاء الله تعالى

اتمنى للجميع حظاً موفقاً وفي رعاية الله ...